

# **ΕΡΓΑΣΤΗΡΙΟ ΗΛΕΚΤΡΟΝΙΚΩΝ**

## **ΕΙΣΑΓΩΓΗ ΣΤΗ ΧΡΗΣΗ ΤΟΥ SPICE/PSpice**

### **ΣΗΜΕΙΩΣΕΙΣ**

Εμμ. Δρης, Καθηγητής, Σ. Μαλτέζος, Επικ. Καθηγητής  
Σχολή Εφαρμοσμένων Μαθηματικών και Φυσικών Επιστημών  
Ε.Μ. Πολυτεχνείο

Αθήνα 2005



## ΕΙΣΑΓΩΓΗ

Ο σχεδιασμός ηλεκτρονικών συστημάτων είναι μια θεμελιώδης διαδικασία, κατά την οποία καλείται κανείς να συνδυάσει κατάλληλα ένα σύνολο ηλεκτρονικών στοιχείων και διατάξεων για να επιτελέσουν μια συγκεκριμένη λειτουργία.

Η κατασκευή ενός «προτύπου» ηλεκτρονικού συστήματος είναι απαραίτητη και για την επιβεβαίωση της καλής λειτουργίας καθώς και την αξιολόγηση των επιδόσεων που επιτυγχάνονται.

Η προσομοίωση είναι αναπόσπαστο μέρος της διαδικασίας σχεδιασμού προτού κατασκευαστεί το πρότυπο κύκλωμα και αφορά στην πλήρη ανάλυση του κυκλώματος υπό συγκεκριμένες συνθήκες και εισόδους, για να διαπιστωθεί αν αυτό συμπεριφέρεται σύμφωνα με τους στόχους που έχουν τεθεί αρχικά. Πιο συγκεκριμένα, ο έλεγχος μέσω της προσομοίωσης, μπορεί να περιλαμβάνει, το σημείο ηρεμίας (dc) (για κυκλώματα με τρανζίστορ), τον έλεγχο απόκρισης του κυκλώματος σε παλμούς ή σήματα ac, την απολαβή (ενίσχυση, κέρδος) ως συνάρτηση της συχνότητας. Συνεπώς, με την προσομοίωση έχουμε τη δυνατότητα να πραγματοποιήσουμε οποιονδήποτε λειτουργικό έλεγχο του κυκλώματος αλλάζοντας παραμέτρους, πράγμα που είναι αδύνατο να γίνει συστηματικά με κατασκευές πολλών προτύπων.

Η επίλυση των θεωρητικών μοντέλων σε κυκλώματα με πολλά στοιχεία ή βαθμίδες είναι περίπλοκη ή και αδύνατη με αναλυτικό τρόπο. Για το λόγο αυτό γίνεται χρήση προγραμμάτων ανάλυσης με υπολογιστή, που είναι αποτελεσματικά και επιτρέπουν τη μελέτη της επίδρασης μεγάλου πλήθους παραγόντων ή παραμέτρων σε πολύ μικρό χρονικό διάστημα. Χρειάζεται αυτό που λέμε, γενικώς, «Computer Aided Circuit Analysis».

Στα μέσα της δεκαετίας του 1960 άρχισε να χρησιμοποιείται η μέθοδος της προσομοίωσης στο σχεδιασμό ή ανάλυση κυκλωμάτων. Τότε, αναπτύχθηκε από την εταιρία IBM το πρόγραμμα ECAP (Electric Circuit Analysis Program). Αυτό που χρειαζόταν να κάνει ο χρήστης, ήταν, να περιγράψει το κύκλωμα με συμβολισμό κόμβων (κομβικός συμβολισμός, nodal notation), που θα τον δούμε παρακάτω. Μετά από αυτό, το πρόγραμμα φρόντιζε για την κατάστρωση των εξισώσεων του κυκλώματος και τη λύση τους με αριθμητικές μεθόδους. Ακόμη και πριν αναπτυχθεί το ECAP γινόταν χρήση υπολογιστών για τη λύση εξισώσεων κυκλωμάτων, αλλά έπρεπε κάθε φορά να γράφεται ειδικό πρόγραμμα για τη συγκεκριμένη εξίσωση του μοντέλου του κυκλώματος. Αυτή η διαδικασία ήταν χρήσιμη αλλά είχε δυσκολίες και εφαρμοζόταν στην πράξη, μόνο σε απλές περιπτώσεις. Μετά από το Πρόγραμμα ECAP αναπτύχθηκαν πολλά παρόμοια προγράμματα τα οποία αποτελούσαν βελτιώσεις του αρχικού. Μερικά από αυτά είναι το SPECTRE, το TRAC, το NET και το CIRCUS. Ένα από αυτά τα προγράμματα είναι και το SPICE (Simulation Program with Integrated Circuit Emphasis). Μια επέκτασή του είναι το PSpice (Personal computer Simulation Program with Integrated Circuit Emphasis). Στο παρόν εγχειρίδιο θα ασχοληθούμε μόνο με τα δύο τελευταία προγράμματα.

## SPICE

Το SPICE, όπως προαναφέραμε, είναι ακρωνύμιο που σημαίνει Simulation Program with Integrated Circuit Emphasis (πρόγραμμα εξομοίωσης με έμφαση στα ολοκληρωμένα κυκλώματα). Αναπτύχθηκε μέσα στη δεκαετία του 1970 από την ομάδα Ολοκληρωμένων Κυκλωμάτων του Ηλεκτρονικού Εργαστηρίου Ερευνών και το Τμήμα Ηλεκτρολόγων Μηχανικών και Υπολογιστών του Πανεπιστημίου του Berkeley της California των Η.Π.Α.

Ο άνθρωπος που πρώτος ανέπτυξε το SPICE είναι ο Dr Lawrence Nagel. Στο διδακτορικό του Nagel περιγράφονται οι αλγόριθμοι και οι αριθμητικές μέθοδοι που χρησιμοποιούνται στο SPICE. Το SPICE από τότε που πρωτοεμφανίστηκε, έχει υποστεί πολλές μεταβολές και βελτιώσεις καθώς βρίσκονται λάθη (bugs), κατά την χρήση του και διορθώνονται. Παρόλο που ο χρήστης δεν είναι αναγκαίο να γνωρίζει τους μαθηματικούς αλγόριθμους που χρησιμοποιούνται στο SPICE, πρέπει να έχει ειδικές γνώσεις σχεδιασμού ηλεκτρικών και ηλεκτρονικών κυκλωμάτων.

Το κύκλωμα περιγράφεται με τη χρήση ενός αρχείου (file) το οποίο αποτελεί το αρχείο εισόδου (αρχείο εισαγωγής δεδομένων) του SPICE. Το αρχείο αυτό περιέχει τα στοιχεία του κυκλώματος (αντιστάτες, πυκνωτές, αυτεπαγωγές, πηγές τάσης και ρεύματος, διατάξεις ημιαγωγών κ.α.) και περιγράφει πως συνδέονται μεταξύ τους κάνοντας χρήση αριθμών που αντιστοιχούν στους διάφορους κόμβους του κυκλώματος. Μπορεί να υπάρχουν επίσης πληροφορίες που δείχνουν τη συχνότητα των πηγών, τη θερμοκρασία, το είδος των αναλύσεων που χρειάζεται να γίνουν και πως πρέπει να παρουσιαστούν τα αποτελέσματα της κάθε ανάλυσης. Το πρόγραμμα SPICE διαβάζει το αρχείο εισόδου και βρίσκει πως πρέπει να συνδεθούν τα διάφορα στοιχεία κυκλωμάτων και διατάξεις σε κάθε κόμβο.

Κατόπιν, για τη λύση του προβλήματος, χρησιμοποιεί την μέθοδο των κόμβων όπου άγνωστοι είναι τα δυναμικά των κόμβων ως προς κάποιον κόμβο αναφοράς (του οποίου το δυναμικό λαμβάνεται ως 0). Το ρεύμα κάθε κλάδου εκφράζεται συναρτήσει της διαφοράς δυναμικού στα άκρα του και της αγωγιμότητας του. Στη συνέχεια χρησιμοποιείται ο νόμος των ρευμάτων του Kirchhoff και κατόπιν καταστρώνεται ένα σύστημα εξισώσεων για το κύκλωμα το οποίο όταν λυθεί θα προσδιορίσει τα δυναμικά (τάσεις) των κόμβων. Προφανώς το SPICE πρέπει να γνωρίζει το μοντέλο των στοιχείων του κυκλώματος και των ημιαγωγικών διατάξεων. Για την λύση του συστήματος των εξισώσεων χρησιμοποιείται η αριθμητική μέθοδος Newton – Raphson.

Υπάρχουν πολλά εμπορικά πακέτα προγραμμάτων που βασίζονται στο SPICE. Πολλά από αυτά περιλαμβάνουν χρήσιμα εργαλεία λογισμικού τα οποία έχουν προστεθεί για να κάνουν ευκολότερη την χρήση του. Εδώ, όπως είπαμε προηγουμένως, θα κάνουμε χρήση του πακέτου PSpice το οποίο είναι μια παραλλαγή του SPICE που ανέπτυξε η εταιρία MicroSim. Το PSpice, εκτός των άλλων, περιλαμβάνει τα προγράμματα Schematics και Probe. Το Schematics χρησιμοποιεί GUI (Graphical User Interface) ώστε μπορεί να σχεδιάζει κανείς στην οθόνη του υπολογιστή το κύκλωμα με τα γνωστά σύμβολα των στοιχείων και με διάφορα μενού (menu) μπορεί να κάνει κάθε αλλαγή εύκολα. Το σχετικό αρχείο είναι τύπου .SCH. Από το κύκλωμα, το PSpice δημιουργεί text files που περιέχουν τις εντολές που περιγράψαμε. Εμείς χρησιμοποιούμε την Φοιτητική Έκδοση που δεν έχει όλες τις δυνατότητες του πλήρους πακέτου.

Το Probe είναι ένα πρόγραμμα που επιτρέπει την απεικόνιση των αποτελεσμάτων στην οθόνη, και μπορεί να ενεργοποιηθεί αν μεταξύ των εντολών υπάρχει η εντολή .PROBE. Με κειμενογράφο μπορούμε να φτιάξουμε text files (τύπου .CIR) που

περιέχει τις κατάλληλες εντολές. Αυτό το αρχείο μπορεί να διαβαστεί από το PSpice το οποίο, με βάση τις εντολές, πραγματοποιεί την επιδιωκόμενη ανάλυση.

## ΔΙΑΔΙΚΑΣΙΑ ΧΡΗΣΗΣ ΤΟΥ SPICE

Ο χρήστης του SPICE ακολουθεί την παρακάτω διαδικασία:

α) Δημιουργεί ένα αρχείο εισόδου (source file) στο οποίο περιγράφει το κύκλωμα για το οποίο θα γίνει ανάλυση ή προσομοίωση.

β) Εισάγει αυτό το αρχείο στο πρόγραμμα το οποίο με τη σειρά του δημιουργεί το αρχείο εξόδου.

γ) Δίνει εντολή να τυπωθεί το αρχείο εξόδου και αν απαιτείται κατασκευάζει διαγράμματα χρησιμοποιώντας αποτελέσματα του αρχείου αυτού.

Το αρχείο εισόδου (τύπου .CIR) μπορεί να γραφτεί με οποιονδήποτε κειμενογράφο, αρκεί να μην εισάγει χαρακτήρες ελέγχου. Το PSpice διαθέτει κατάλληλο κειμενογράφο που καλό είναι να προτιμάται. Το αρχείο εισόδου περιέχει, εντολές δεδομένων που περιγράφουν το κύκλωμα, περιέχει εντολές ελέγχου οι οποίες καθορίζουν το είδος της ανάλυσης και επίσης εντολές εξόδου οι οποίες περιγράφουν το είδος και τον τρόπο παρουσίασης των αποτελεσμάτων. Για να φτιάξουμε στο αρχείο εισόδου τις εντολές δεδομένων, αρχίζουμε με ένα διάγραμμα του κυκλώματος (schematic) και σημειώνουμε σε αυτό, σύμφωνα με τους κανόνες του SPICE, ορισμένα στοιχεία που περιγράφονται στις παρακάτω τρεις εργασίες:

1) Δίνουμε σε κάθε στοιχείο του κυκλώματος ένα όνομα. Το αρχικό γράμμα του ονόματος πρέπει να είναι σύμφωνο με τον Πίνακα 1. Στο PSpice τα γράμματα μπορεί να είναι πεζά και κεφαλαία. Το πρόγραμμα SPICE, δηλαδή το πακέτο λογισμικού SPICE2G.6 του Πανεπιστημίου Berkeley χρειάζεται κεφαλαία. Στο PSpice τα RLOAD ή rbaseq1 παριστάνουν αντιστάτες.

2) Ορίζουμε ένα «κόμβο αναφοράς», κόμβος 0. Αυτός ονομάζεται και «κόμβος γείωσης» και θεωρείται ότι έχει δυναμικό 0. Τα δυναμικά όλων των άλλων κόμβων υπολογίζονται ως προς αυτόν. Ο κόμβος αναφοράς μπορεί να μην είναι η πραγματική γείωση του κυκλώματος αλλά απλώς ο κόμβος αναφοράς για τους υπολογισμούς.

3) Δίνουμε στους άλλους κόμβους του κυκλώματος έναν αριθμό (κομβικός αριθμός) που πρέπει να είναι θετικός ακέραιος. Η διάταξη στην αρίθμηση των κόμβων είναι αυθαίρετη και οι αριθμοί δεν είναι κατ'ανάγκη διαδοχικοί. Ο κάθε κόμβος πρέπει να συνδέει τουλάχιστον δύο στοιχεία του κυκλώματος.

Στο αρχείο εισόδου χρησιμοποιούμε εντολές ελέγχου που καθορίζουν τι είδους ανάλυση θέλουμε να κάνουμε. Το SPICE γενικώς μπορεί να κάνει ανάλυση για ac, dc, μεταβατικά φαινόμενα (transient), συνάρτηση μεταφοράς dc, ευαισθησία σε μικρά σήματα dc, παραμόρφωση, θόρυβο και ανάλυση Fourier. Για το σκοπό αυτό χρειάζεται να εισαχθούν στο αρχείο εισόδου μερικές (απλές) εντολές ελέγχου. Μια τέτοια εντολή μπορεί να καθορίζει την τιμή της τάσης ή ρεύματος κάποιας πηγής, το εύρος συχνοτήτων που πρέπει να έχει μία πηγή ac, το χρονικό διάστημα για το οποίο θα γίνει η ανάλυση μεταβατικών φαινομένων (μεταβατική ανάλυση) και το βήμα χρόνου που θα χρησιμοποιηθεί. Άλλες εντολές ελέγχου καθορίζουν τον τρόπο παρουσίασης των αποτελεσμάτων της ανάλυσης, π.χ. σε μορφή πίνακα ή σε γράφημα.

Πίνακας 1	
C	Χωρητικότητα (πυκνωτής)
D	Δίοδος
E	Πηγή τάσης ελεγχόμενη από τάση
F	Πηγή ρεύματος ελεγχόμενη από ρεύμα
G	Πηγή ρεύματος ελεγχόμενη από τάση
H	Πηγή τάσης ελεγχόμενη από ρεύμα
I	Ανεξάρτητη πηγή ρεύματος
J	Τρανζίστορ επίδρασης-πεδίου διεπαφής (JFET)
K	Συντελεστής σύζευξης αμοιβαίας επαγωγής
L	Αυτεπαγωγή
M	Τρανζίστορ επίδρασης – πεδίου μετάλλου-οξειδίου (MOSFET)
Q	Διπολικό τρανζίστορ διεπαφής (BJT)
T	Γραμμή μεταφοράς
V	Ανεξάρτητη πηγή τάσης

Η έξοδος του SPICE αποτελείται από πολλά μέρη. Οι εντολές εξόδου στο αρχείο εισόδου ρυθμίζουν αυτά τα μέρη και τον τρόπο εμφάνισης των αποτελεσμάτων. Στο πρώτο μέρος του αρχείου εξόδου βρίσκεται η περιγραφή του κυκλώματος. Κατ' ελάχιστο, αυτό το μέρος περιλαμβάνει τη λίστα των εντολών του αρχείου εισόδου που φτιάχτηκε για το SPICE για να χρησιμοποιηθεί στην ανάλυση.

Το δεύτερο μέρος περιέχει την προκαθορισμένη (default) έξοδο για κάποιες από τις εντολές ανάλυσης που δημιουργούν έξοδο χωρίς οδηγίες από το αρχείο εισόδου. Για παράδειγμα η εντολή εξόδου `.TF` πάντα οδηγεί στο να γραφούν στο αρχείο εξόδου η απολαβή (ένισχυση), η αντίσταση εισόδου και η αντίσταση εξόδου αφού ολοκληρωθεί η ανάλυση.

Το τρίτο μέρος σχετίζεται με την εκτύπωση των αποτελεσμάτων και γραφικών παραστάσεων που έχουν ζητηθεί στο αρχείο εισόδου. Η εντολή `.PRINT` οδηγεί στην εκτύπωση δεδομένων της ανάλυσης. Η εντολή `.PLOT` οδηγεί στο να γίνουν γραφήματα στον εκτυπωτή.

Το τέταρτο μέρος περιέχει πληροφορίες που συνοψίζουν μερικές στατιστικές πληροφορίες που περιγράφουν την εκτέλεση του PSpice. Τέτοιες είναι οι απαιτήσεις για μνήμη και ο χρόνος εκτέλεσης που είναι χρήσιμα στοιχεία που μπορούν, σε συνδυασμό με προαιρετικές παραμέτρους, να κάνουν την ανάλυση πιο αποτελεσματική.

Επιπλέον, το πακέτο PSpice περιλαμβάνει ένα graphics postprocessor (μεταεπεξεργαστή γραφικών) που λέγεται PROBE. Το PROBE μας επιτρέπει να κάνουμε γραφήματα πρακτικώς κάθε φυσικού μεγέθους που μας ενδιαφέρει, μετά από την ανάλυση του κυκλώματος από το SPICE. Το PROBE χρειάζεται να έχει πρόσβαση σε ειδικό αρχείο για να λειτουργήσει σωστά. Ένα τέτοιο αρχείο δημιουργείται κατά τη διάρκεια της ανάλυσης με το PSpice αν συμπεριληφθεί η εντολή `.PROBE` στο αρχείο εισόδου. Αφού τελειώσει η ανάλυση μπορεί να ενεργοποιηθεί το PROBE στο επίπεδο του λειτουργικού συστήματος ή στο επίπεδο του κελύφους οδηγιών (command shell) όπως θα δούμε παρακάτω.

Συνοψίζοντας, η διαδικασία που ακολουθούμε είναι: Δημιουργούμε το αρχείο εισόδου για το SPICE χρησιμοποιώντας έναν επεξεργαστή κειμένου (text editor). Το

PSpice έχει τον δικό του επεξεργαστή κειμένου. Το αρχείο είναι τύπου .CIR. Χρειάζεται πάντα στην πρώτη γραμμή να υπάρχει ο τίτλος. Η τελευταία γραμμή είναι το .END . Η διάταξη των άλλων γραμμών του κώδικα του προγράμματος μπορεί να είναι οποιαδήποτε. Φυσικά όταν υπάρχουν υποκυκλώματα (που είναι κυκλώματα που καλούνται όπως οι υπορουτίνες) ο κανόνας αυτός έχει τους προφανείς περιορισμούς. Τρέχουμε το PSpice το οποίο διαβάζει το αρχείο εισόδου, κάνει την ανάλυση του κυκλώματος και δημιουργεί το αρχείο εξόδου τύπου .OUT . Μπορούμε να έχουμε και γραφήματα διαφόρων ηλεκτρικών μεγεθών που αφορούν το κύκλωμα.

- Αριθμητικές τιμές.

Το SPICE αναγνωρίζει αριθμούς εκφρασμένους με διάφορους τρόπους. Μπορεί να χρησιμοποιηθεί ο επιστημονικός συμβολισμός (scientific notation) 5.23E04, 8.45E-6 κτλ, όπου E σημαίνει δύναμη του 10. Επίσης μπορεί να χρησιμοποιηθούν τα προθέματα του Πίνακα 2.

Σημειώνουμε ότι, παρόλο που σε όλες τις γλώσσες εκτός από την αγγλική το σύμβολο των δεκαδικών είναι το κόμμα σε αυτά τα πακέτα (δυστυχώς) χρησιμοποιείται παντού μόνο η «κάτω» τελεία. Επίσης ο συμβολισμός των προθεμάτων δεν είναι ακριβώς αυτός του Διεθνούς Συστήματος μονάδων (SI). Τα γράμματα που δεν ανήκουν στον Πίνακα 2 αγνοούνται. Επίσης, αγνοούνται τα γράμματα που ακολουθούν μετά από ένα έγκυρο πρόθεμα.

Π.χ.

270V σημαίνει 270 (αγνοείται το V)

0.27KW σημαίνει 270 (μεταφράζεται το K ως  $1 \times 10^3$  και το W αγνοείται)

0.27KOHM σημαίνει επίσης 270 (πάλι το K μεταφράζεται ως  $1 \times 10^3$  και τα υπόλοιπα αγνοούνται).

Πίνακας 2		
Σύμβολο	Συντελεστής	Πρόθεμα
F	$1 \times 10^{-15}$	femto
P	$1 \times 10^{-12}$	pico
U	$1 \times 10^{-6}$	micro
M	$1 \times 10^{-3}$	milli
MIL	$25,4 \times 10^{-6}$	
K	$1 \times 10^3$	kilo
MEG	$1 \times 10^6$	mega
G	$1 \times 10^9$	giga
T	$1 \times 10^{12}$	tera

- Κανόνες για τις «γραμμές» (εντολές) του αρχείου εισόδου.

Κάθε αλφαριθμητική (γραμματαριθμητική) σειρά που παριστάνει στοιχείο κυκλώματος, αρχίζει με τα γράμματα του Πίνακα 1 . Στο SPICE οι επιπλέον των 8

χαρακτήρες αγνοούνται ενώ στο PSpice μπορεί να υπάρχουν μέχρι 131 χαρακτήρες. Συνιστάται φυσικά, να γίνεται χρήση μικρού (λογικού) πλήθους χαρακτήρων. Η συνέχιση μιας γραμμής δηλώνεται με το + στην πρώτη στήλη της επόμενης γραμμής. Το σύμβολο \* στην αρχή της γραμμής σημαίνει επεξήγηση (comment). Ό,τι γράφεται μεταξύ < > είναι προαιρετικό.

## Αντιστάτης

Rxxxxxx N1 N2 VALUE <TC=TC1<,TC2>>

Αυτό παριστάνει τον αντιστάτη Rxxxxxx συνδεδεμένο μεταξύ των κόμβων N1 και N2. Η ονομαστική τιμή αντίστασης είναι η value, στην ονομαστική τιμή θερμοκρασίας ( $t_{nom}$ ). Η τιμή της αντίστασης στη θερμοκρασία temp δίνεται από τον τύπο:

$$VALUE(temp) = VALUE(t_{nom}) * (1 + TC1 * (temp - t_{nom}) + TC2 * (temp - t_{nom})^2)$$

Δηλαδή τα TC1 και TC2 είναι συντελεστές στη σχέση δευτέρου βαθμού που καθορίζει την εξάρτηση της αντίστασης από τη θερμοκρασία. Αν τα TC1 TC2 δεν δοθούν, θεωρούνται μηδέν.

Παραδείγματα:

Απλή περίπτωση:

R1 5 11 3.6E3

Αυτό σημαίνει ότι ο αντιστάτης R1 είναι συνδεδεμένος μεταξύ των κόμβων 5 και 11 και έχει τιμή  $3,6 \times 10^3 \Omega = 3600 \Omega$ .

## Πυκνωτής

Cxxxxxx N+ N- VALUE <IC= INCOND >

N1, N2 είναι οι κόμβοι όπου είναι συνδεδεμένος ο πυκνωτής. N+ είναι ο θετικός κόμβος και N- ο αρνητικός. IC είναι η τιμή τάσης στον πυκνωτή την χρονική στιγμή μηδέν και είναι προαιρετική.

Αν έχει δοθεί τέτοια αρχική συνθήκη, λαμβάνεται υπόψη στην ανάλυση μόνο αν έχει συμπεριληφθεί στην εντολή ελέγχου .TRAN η δυνατότητα UIC (use initial conditions, χρησιμοποίησε αρχικές συνθήκες). Η εντολή .TRAN σημαίνει ανάλυση μεταβατικών φαινομένων (transient), μεταβατική ανάλυση.

Παράδειγμα:

C15 31 7 15U IC=-1V



Αυτό σημαίνει ότι ο πυκνωτής C15, είναι συνδεδεμένος μεταξύ του κόμβου 31 (θετικός ακροδέκτης) και του κόμβου 7 (αρνητικός ακροδέκτης). Η χωρητικότητά του είναι  $15 \cdot 10^{-6} \text{ F} = 15 \text{ }\mu\text{F}$  και ο πυκνωτής είναι αρχικά φορτισμένος σε τάση 1 V με τον κόμβο 31 αρνητικό ως προς τον 7.

Αν ο πυκνωτής δεν είναι γραμμικός και η χωρητικότητά του εξαρτάται από την τάση του, αυτό δηλώνεται με την εντολή:

Cxxxxxx N+ N- POLY C0 C1 C2... <IC=INCOND>

όπου C0, C1, C2... συντελεστές στο πολυώνυμο:  $C = C0 + C1 \cdot V + C2 \cdot V^2 + \dots$  Το C είναι σε farad και το V σε volt.

### Αυτεπαγωγή

Lxxxxxx N+ N- VALUE <IC=INCOND>

Τα N+ , N- είναι όπως πριν, το προαιρετικό IC δηλώνει το αρχικό ρεύμα. Όταν το IC είναι θετικό σημαίνει ότι έχει φορά από το N+ προς το N-.

Για να χρησιμοποιηθεί η αρχική συνθήκη πρέπει να έχει ενεργοποιηθεί το UIC στο .TRAN.

Παράδειγμα:

L11 13 85 2.37 IC=-5

Ο παραπάνω συμβολισμός δηλώνει ότι η αυτεπαγωγή L11 είναι συνδεδεμένη μεταξύ των κόμβων 13 και 85 έχει τιμή 2.37 henry, χένρυ (H) και διαρρέεται από αρχικό ρεύμα 5 A από τον κόμβο 85 προς τον κόμβο 13.

Αν η αυτεπαγωγή δεν είναι γραμμική αλλά εξαρτάται από το ρεύμα τότε έχουμε,

Lxxxxxx N+ N- POLY L0 L1 L2... <IC=INCOND>

Όπου L0 L1 L2... , οι συντελεστές του πολυωνύμου  $L = L0 + L1 \cdot I + L2 \cdot I^2 + L3 \cdot I^3 + \dots$

Το ρεύμα είναι σε ampere (A) και το L σε henry (H).

### Αμοιβαία επαγωγή

Kxxxxxx Lyyyyyy Lzzzzzz VALUE

Αυτό σημαίνει ότι η αμοιβαία επαγωγή εξαιτίας της σύζευξης μεταξύ των αυτεπαγωγών Lyyyyyy και Lzzzzzz έχει συντελεστή σύζευξης με τιμή ίση με VALUE (ισχύει  $0 < \text{VALUE} \leq 1$ ).

Δεν χρειάζονται αριθμοί για τους κόμβους διότι αυτοί υπάρχουν στις «δηλώσεις» των Lyyyyyy και Lzzzzzz.

Προφανώς Ισχύει  $M = K \sqrt{L_1 L_2}$

Παράδειγμα:

LP 10 16 3.5E-6  
LS 13 18 1.6E-6  
KPRISEC LP LS 0.97

Ο συντελεστής σύζευξης είναι 0,97 .

Ανεξάρτητες πηγές.

- Για πηγή τάσης έχουμε:

Vxxxxxx N+ N- < <DC>DC/TRAN-VALUE> <AC<ACMAG<ACPHASE>>>  
<TRANKIND (TPAR1 TPAR2...)>>

- Για πηγή ρεύματος έχουμε:

Ixxxxxx N+ N- < <DC>DC/TRAN-VALUE> <AC<ACMAG<ACPHASE>>>  
<TRANKIND (TPAR1 TPAR2...)>>

Το N+ και το N- σημαίνει ότι θετικό ρεύμα ρέει από τον κόμβο N+ προς τον κόμβο N- μέσα από την πηγή. Τα άλλα σύμβολα είναι γενικά προαιρετικά. Το DC μετά τα N+ και N- σημαίνει ότι η τάση είναι σταθερή με το χρόνο.

Το DC/TRAN-VALUE καθορίζει το μέγεθος (VALUE) της πηγής για την οποία θα γίνει ανάλυση DC και μεταβατική ανάλυση. Αν είναι μηδέν μπορεί να μην δοθεί καθόλου.

Το AC σημαίνει αρμονική εξάρτηση με τον χρόνο. Το πλάτος και η φάση σε αυτή την περίπτωση είναι ACMAG και ACPHASE (σε μοίρες).

Τα ACMAG και ACPHASE μπορεί να παραληφθούν οπότε το ACMAG έχει την προκαθορισμένη τιμή 1 V και το ACPHASE την τιμή μηδέν.

Επιπλέον, μόνο για μεταβατική ανάλυση, μία ανεξάρτητη πηγή μπορεί να περιγράφεται με μία από τις 5 συναρτήσεις του χρόνου, 5 δυνατότητες, (TRANKIND): pulse, sinusoidal, exponential, piece wise linear ή μοναδικής συχνότητας FM.

Τα TPAR1, TPAR2,... είναι παράμετροι σχετικές με την ειδική πηγή TRANKIND. Αυτές οι πηγές περιγράφονται λίγο αργότερα. Αν για κάποια πηγή δεν δοθεί καμία από τις 5 δυνατότητες τότε η πηγή θεωρείται νεκρή πηγή. Αυτή είναι χρήσιμη όπως θα δούμε παρακάτω (VBORING). Το SPICE θεωρεί ότι το ρεύμα ρέει διαμέσου της πηγής από το N+ προς το N- .

Παραδείγματα:

VBAT1 11 17 DC 5.

Δηλαδή, έχουμε την VBAT1 η οποία είναι ανεξάρτητη πηγή τάσης DC μεταξύ των κόμβων 11 και 17. Ο κόμβος 11 έχει θετικό δυναμικό 6 V ως προς τον κόμβο 17.

IL1 32 40 SIN(15M 30M 25K)

Αυτό σημαίνει ανεξάρτητη πηγή ρεύματος μεταξύ κόμβων N+ = 32 και N- =40, με συνιστώσα DC (offset, απόκλιση) ίση με 15 mA και με συνιστώσα αρμονική με το χρόνο (όταν γίνεται μεταβατική ανάλυση), με μέγιστο πλάτος 30 mA και συχνότητα 25 kHz.

VBORING 3 6

Αυτή είναι μία νεκρή πηγή τάσης μεταξύ των κόμβων 3 και 6. Συμπεριφέρεται σαν βραχυκύκλωμα. Χρησιμοποιείται ως αμπερόμετρο που μπορεί να τοποθετηθεί σε οποιοδήποτε κλάδο. Το θετικό ρεύμα ρέει από τον κόμβο 3 προς τον 6 μέσα από την πηγή.

VFM1 13 10 AC 2 30 LSFFM(0 3 1K 3.5 300)

Σημαίνει ότι η πηγή τάσης VFM1 μπορεί να είναι ένα από τα δύο πράγματα, ανάλογα με την ανάλυση που γίνεται. Συγκεκριμένα, αν γίνεται ανάλυση AC (με την βοήθεια της εντολής ελέγχου .AC), τότε η VFM1 είναι πηγή αρμονική με τον χρόνο πλάτους 2 V και (αρχικής) φάσης 30°. Η συχνότητα καθορίζεται με την εντολή .AC .

Αν γίνεται μεταβατική ανάλυση (με χρήση της εντολής ελέγχου .TRAN), η VFM1 γίνεται πηγή FM (Frequency Modulation, διαμόρφωση συχνότητας) με φέρουσα συχνότητα 1 kHz, δείκτη διαμόρφωσης 3,5 και συχνότητα διαμόρφωσης (σήμα) 300 Hz.

VSYN3 3 8 PULSE(-7 10 10M 1U 2U 12M 30M)

Εδώ έχουμε μία ομάδα (τρένο) παλμών που αρχίζει με -7 V και φτάνει στα 10 V, ξεκινά με καθυστέρηση 10 ms από την στιγμή t=0, έχει χρόνο ανόδου 1 μs και χρόνο καθόδου 2 μs, έχει εύρος παλμού 12 ms και περίοδο 30 ms.

ITRIANGL 18 37 PWL(0 0 0.25 3 0.75 -3 1.25 3 1.75 -3 2 0)

Αυτό περιγράφει δύο κύκλους μίας τριγωνικής κυματομορφής ρεύματος που ξεκινά από 0 A, και ανεβοκατεβαίνει μεταξύ +3 A και -3 A, με περίοδο 1 s. Το ρεύμα της πηγής ITRIANGL είναι θετικό από τον κόμβο 18 προς τον 37 διαμέσου της πηγής.

- Συναρτήσεις του χρόνου για ανεξάρτητες πηγές.

Υπάρχουν 5 συναρτήσεις του χρόνου για μεταβατική ανάλυση με χρήση ανεξάρτητων πηγών:


1. PULSE. Ένας μοναδικός παλμός ή τρένο παλμών με χρόνο ανόδου και χρόνο καθόδου που μπορούν να καθοριστούν.

Vxxxxxx } N+ N- PULSE (V1 V2 TD TR TF PW PER)  
Ixxxxxx }

Αυτή η εντολή δείχνει τη χρήση του PULSE για ανεξάρτητες πηγές τάσης και ρεύματος.

V1 αρχική τιμή σε volt ή ampere (η τιμή πρέπει να καθοριστεί), V2 τελική τιμή σε volt ή ampere (η τιμή πρέπει να καθοριστεί), TD Χρονική καθυστέρηση (η προκαθορισμένη τιμή, default, είναι 0) σε second, TR χρόνος ανόδου (η προκαθορισμένη τιμή είναι tstep) σε second, TF χρόνος καθόδου (η προκαθορισμένη τιμή είναι tstep) σε second, PW εύρος παλμού (η προκαθορισμένη τιμή είναι tstop) σε second, PER περίοδος (η προκαθορισμένη τιμή είναι tstop) σε second.

Τα TSTEP και TSTOP αναφέρονται στα μεγέθη του βήματος και τελικού χρόνου για την μεταβατική ανάλυση και καθορίζονται με την εντολή .TRAN TSTEP TSTOP . Βλέπε Σχήμα 1 .




Σχήμα 1

2. SIN. Αρμονική με τον χρόνο εξάρτηση όπου το ξεκίνημα μπορεί να γίνει μετά από καθυστέρηση από το μηδέν. Μπορεί να έχει και εξάρτηση φθίνουσα (αρμονική) με τον χρόνο.

Η χρήση για ανεξάρτητες πηγές είναι:

Vxxxxxx } N+ N- SIN(V0 VA FREQ TD THETA)  
Ixxxxxx }

V0 απόκλιση (offset) σε volt ή ampere (η τιμή πρέπει να καθοριστεί), VA πλάτος σε volt ή ampere (η τιμή πρέπει να καθοριστεί), FREQ συχνότητα (η προκαθορισμένη τιμή είναι 1/TSTOP) σε Hz, TD καθυστέρηση (η προκαθορισμένη τιμή είναι 0) σε second, THETA συντελεστής απόσβεσης (η προκαθορισμένη τιμή είναι 0) σε 1/s. Βλέπε Σχήμα 2.



Σχήμα 2

Τα TSTEP και TSTOP βρίσκονται στην εντολή .TRAN TSTEP TSTOP και είναι το βήμα και ο μέγιστος χρόνος της μεταβατικής ανάλυσης.

Γενικά ισχύει για  $t \leq TD$  ισχύει  $V = V_0$   
και για  $t \geq TD$  έχουμε

$$V = V_0 + V_A * \exp(-(t-TD) * \thetaeta) * \sin(2 * \pi * \text{freq} * (t - TD))$$

3. EXP Μοναδικός παλμός με άνοδο εκθετική με το χρόνο και κάθοδο εκθετική με το χρόνο.

Για ανεξάρτητες πηγές έχουμε,

$$\left. \begin{array}{l} V_{xxxxxx} \\ I_{xxxxxx} \end{array} \right\} N+ \ N- \ EXP(V1 \ V2 \ TD1 \ TAU1 \ TD2 \ TAU2)$$

Οι μονάδες χρόνου είναι σε second. Υπάρχουν οι ακόλουθες προκαθορισμένες τιμές :  
TD1=0, TAU1=TSTEP, TD2=TD1+TSTEP, TAU2=TSTEP.

Τα V1 και V2 πρέπει να καθοριστούν.

Το TSTEP είναι το βήμα στην μεταβατική ανάλυση και καθορίζεται με την εντολή,  
.TRAN TSTEP TSTOP

Έχουμε για  $0 \leq t \leq TD1$   $V_{xxxxxx} = V1$ .


Για  $TD1 \leq t \leq TD2$  έχουμε

$$\left. \begin{array}{l} V_{xxxxxx} \\ I_{xxxxxx} \end{array} \right\} = V_1 + (V_2 - V_1) * (1 - \exp(-(t - TD_1) / TAV_1))$$

Για  $TD_2 < t < TSTOP$  έχουμε,

$$\left. \begin{array}{l} V_{xxxxxx} \\ I_{xxxxxx} \end{array} \right\} = V_1 + (V_2 - V_1) * (1 - \exp(-(t - TD_1) / TAV_1)) + (V_1 - V_2) * (1 - \exp(-(t - TD_2) / TAV_2))$$

Βλέπε Σχήμα 3 .



Σχήμα 3

4. PWL Κατά τμήματα γραμμική συνάρτηση για ανεξάρτητες πηγές.


$$\left. \begin{array}{l} V_{xxxxxx} \\ I_{xxxxxx} \end{array} \right\} = N+ \quad N- \quad PWL(T1 \quad V1 \quad <T2 \quad V2 \quad T3 \quad V3...>)$$

Οι παράμετροι είναι: Tn (n-οστή τιμή χρόνου) που πρέπει να καθοριστεί και είναι σε second. Vn (n-οστή τιμή τάσης ή ρεύματος) που πρέπει να καθοριστεί και είναι σε volt ή ampere.

Παράδειγμα:

VPWL 1 0 PWL(0 0 2 3 3 2 4 2 4.2 5 4.4 5 4.3 -2 7 1 9 -1 9.5 4 10 3)

Βλέπε Σχήμα 4 .



Σχήμα 4

5. SFFM Διαμόρφωση συχνότητας, FM, (frequency modulation) με μοναδική συχνότητα διαμόρφωσης (σήμα).

$$\left. \begin{array}{l} V_{xxxxxx} \\ I_{xxxxxx} \end{array} \right\} = N+ \quad N- \quad DFFM(V0 \quad VA \quad FC \quad MDI \quad FS)$$

Γενικά ισχύει:

$$\left. \begin{array}{l} V_{xxxxxx} \\ I_{xxxxxx} \end{array} \right\} = V0 + VA * ((2 * \pi * FC * t) + MDI * \sin(2 * \pi * FS * t))$$

Το V0 είναι η απόκλιση (offset) σε volt ή ampere και πρέπει να καθοριστεί. Το VA είναι το πλάτος σε volt ή ampere και πρέπει να καθοριστεί. FC είναι η συχνότητα του φέροντος που έχει ως προκαθορισμένη τιμή την 1/TSTOP σε Hz. MDI είναι ο δείκτης διαμόρφωσης και πρέπει να καθοριστεί. FS είναι η συχνότητα διαμόρφωσης (σήματος) σε Hz με προκαθορισμένη τιμή 1/TSTOP.

## Γραμμικές Ελεγχόμενες πηγές

- Πηγή ρεύματος ελεγχόμενη από ρεύμα

(VCCS voltage controlled current source)

Η σχετική εντολή είναι,


Gxxxxxx N+ N- NC+ NC- VALUE

Τα N+ N- δείχνουν τους κόμβους που είναι συνδεδεμένη η πηγή, με φορά ρεύματος εντός της πηγής από το N+ προς το N-. NC+ NC- είναι οι κόμβοι όπου συνδέεται η ελέγχουσα τάση της οποίας ο θετικός πόλος είναι το NC+. VALUE είναι η τιμή της διαγωγιμότητας σε siemens.

Ισχύει,

$$I=G*V$$

Το V είναι σε volt, το I σε ampere και G σε siemens. Βλέπε Σχήμα 5.



$$Gxxxxxx = VALUE * (V(NC_+) - V(NC_-))$$

Σχήμα 5

- Πηγή τάσης ελεγχόμενη από τάση

(VCVS voltage controlled voltage source).

Η εντολή αρχίζει με το σύμβολο E.

Exxxxxx N+ N- NC+ NC- VALUE

VALUE είναι η ενίσχυση τάσης (πολλαπλασιαστική σταθερά).

Τα άλλα σύμβολα σημαίνουν αυτά που είπαμε προηγουμένως.


Exxxxxx παριστάνει την τάση της ελεγχόμενης πηγής.

Ισχύει

$$V=E*V_{\text{control}}$$

Exxxxxx = VALUE \*(V(NC+) - V(NC-)) . Βλέπε Σχήμα 6.






Σχήμα 6

Παράδειγμα:

Ο τελεστικός ενισχυτής σε σχετικά χαμηλές συχνότητες και μικρά σήματα είναι ένα χαρακτηριστικό παράδειγμα VCVS. Βλέπε Σχ. 7.



Σχήμα 7

Έχουμε τις εντολές:

R1 4 11 50

R2 19 0 1K

R3 22 19 20K

RIN 11 19 1MEG

VS 4 0 AC 1 0

E1 22 0 11 19 5E4

Χρειάζεται η εντολή ελέγχου .AC για τον καθορισμό της συχνότητας (συχνοτήτων) της πηγής VS.

- Πηγή ρεύματος ελεγχόμενη από ρεύμα.  
(CCCS current controlled current source)

Η εντολή είναι:

Fxxxxxx N+ N- VNAME VALUE

VNAME είναι το όνομα της πηγής τάσης μέσα στην οποία ρέει το ρεύμα που ελέγχει την CCCS, VALUE είναι η ενίσχυση ρεύματος της CCCS.


Μπορεί να χρειάζεται να εισαχθεί νεκρή πηγή τάσης στη θέση που ρέει το ελέγχον ρεύμα για να υπάρχει το VNAME . Μια τέτοια πηγή δεν έχει επίδραση στη λειτουργία του κυκλώματος.

Ισχύει,

$$Fxxxxxx = VALUE * (\text{ρεύμα δια μέσου της VNAME})$$

Παράδειγμα:

Ένα διπολικό τρανζίστορ (BJT) ηρη, με γειωμένη βάση, είναι συνδεδεμένο όπως στο Σχήμα 8 .



Σχήμα 8

Έστω ότι η ενίσχυση ρεύματος είναι  $\beta=80$  ( $=I_C/I_B$ )

Έχουμε τις εντολές

```
VIB 0 2 0
FBJT 4 5 VIB 80
VCC 3 0 15
VEE 6 0 -12
RBASE 2 5 1.8K
RE 5 6 10K
RC 3 4 10K
```

- **Πηγή τάσης ελεγχόμενη από ρεύμα**  
(CCVS current controlled voltage source)

Hxxxxxx N+ N- VNAME VALUE

VNAME είναι το όνομα της πηγής τάσης δια της οποίας διέρχεται το ρεύμα που ελέγχει την CCVS.


VALUE είναι η διαντίσταση του CCVS σε ohm.

Μπορεί να χρειαστεί να εισαχθεί μία νεκρή πηγή τάσης στο κύκλωμα στον κλάδο όπου ρέει το ρεύμα ελέγχου ώστε να υπάρχει το VNAME για τον έλεγχο του CCVS. Προφανώς μία τέτοια πηγή δεν επηρεάζει τη λειτουργία του κυκλώματος.

Hxxxxxx = VALUE \* (ρεύμα δια της VNAME)

Βλέπε Σχήμα 9 , για το οποίο έχουμε:

H1 24 37 V1 6



Σχήμα 9

## ΜΗ ΓΡΑΜΜΙΚΕΣ (ΠΟΛΥΩΝΥΜΙΚΕΣ) ΕΛΕΓΧΟΜΕΝΕΣ ΠΗΓΕΣ

- Πολυωνυμικές σχέσεις.

Αν μια πολυωνυμική πηγή εξαρτάται από μόνο μια ελέγχουσα πηγή τότε λέμε ότι είναι μονοδιάστατη, αν εξαρτάται από N πηγές τότε είναι N-διάστατη.

Για τη περίπτωση μιας διάστασης έχουμε το πολυώνυμο,

$$f_v = p_0 + p_1 * f_a + p_2 * f_a^2 + p_3 * f_a^3 + p_4 * f_a^4 + \dots$$

$f_v$  είναι το ελεγχόμενο μέγεθος και  $f_a$  το ελέγχον.

Αν δίνεται μόνον ένας συντελεστής τότε θεωρείται ότι αυτός είναι ο  $p_1$  και θεωρείται ότι  $p_0 = 0$ .

Για δυο διαστάσεις (δυο ελέγχουσες πηγές) έχουμε τη σχέση,

$$f_v = p_0 + p_1 * f_a + p_2 * f_b + p_3 * f_a^2 + p_4 * f_a * f_b + p_5 * f_b^2 + p_6 * f_a^3 + p_7 * f_a^2 * f_b + p_8 * f_a * f_b^2 + p_9 * f_b^3 + \dots$$

Προφανώς οι ελέγχουσες πηγές είναι  $fa, fb$ .

Για τρεις διαστάσεις έχουμε τρεις πηγές τις  $fa, fb, fc$  και τη σχέση,

$$F_v = p_0 + p_1 * fa + p_2 * fb + p_3 * fc + p_4 * fa^2 + p_5 * fa * fb + p_6 * fa * fc + p_7 * fb^2 + p_8 * fb * fc + p_9 * fc^2 + p_{10} * fa^3 + p_{11} * fa^2 * fb + p_{12} * fa^2 * fc + p_{13} * fa * fb^2 + p_{14} * fa * fb * fc + p_{15} * fa * fc^2 + p_{16} * fb^3 + p_{17} * fb^2 * fc + p_{18} * fb * fc^2 + p_{19} * fc^3 + p_{20} * fa^4 + \dots$$

Για περισσότερες διαστάσεις η σχέση είναι αρκετά πολύπλοκη!

- **Μη γραμμικές πηγές ρεύματος ελεγχόμενες από τάση.**

Η γενική μορφή εντολής για μη γραμμική πηγή ρεύματος ελεγχόμενης από τάση (VCCS) είναι,

Gxxxxxx N+ N- <POLY(ND)> NC1+ NC1- <NC2+ NC2->...  
+ P0 P1 ...<IC=...>

N+ , N- είναι οι κόμβοι στους οποίους είναι συνδεδεμένη η πηγή. Το POLY(ND) πρέπει να καθοριστεί αν η VCCS έχει περισσότερες από μια διάσταση, ND είναι η διάσταση (η προκαθορισμένη τιμή είναι POLY(1), δηλαδή μια διάσταση). NC+ , NC- είναι οι θετικός και αρνητικός ελέγχοντες κόμβοι. Η αρχική συνθήκη, IC = ... , Καθορίζει την τιμή των ελεγχουσών τάσεων τη χρονική στιγμή μηδέν. Αν δεν δίνονται αρχικές συνθήκες η προκαθορισμένη τιμή είναι μηδέν.

Παραδείγματα:

A) GSIMPLE 1 4 7 2 0 0.5 0.3 0.06

Το ελεγχόμενο ρεύμα που ρέει από τον κόμβο 1 προς τον κόμβο 2 μέσα από τη πηγή GSIMPLE είναι συνάρτηση της τάσης μεταξύ του κόμβου 7 και του κόμβου 2 και έχει την ακόλουθη εξάρτηση,

$$I = 0 + 0.5 * V(7,2) + 0.3 * V(7,2)^2 + (6E-2) * V(7,2)^3 \quad (\text{η τάση είναι σε } V \text{ και το ρεύμα σε } A).$$

Αφού δεν έχει δοθεί POLY(ND), η πηγή GSIMPLE είναι συνάρτηση μιας μόνο τάσης.

B) GGOLLY 5 9 POLY(2) 1 0 3 4 0.001 7M 80U

Η πηγή ρεύματος GGOLLY που ελέγχεται από τάση, είναι συνδεδεμένη στους κόμβους 5 και 9 ( το θετικό ρεύμα ρέει από τον 5 προς τον 9 μέσα από την GGOLLY) και ελέγχεται από δυο τάσεις. Η πρώτη ελέγχουσα πηγή είναι η τάση του κόμβου 1 ως προς τον κόμβο 0, και η δεύτερη είναι η τάση του κόμβου 3 ως προς τον

κόμβο 4.  $P0 = 1E-3$ ,  $P1 = 7E-3$  και  $P2 = 8E-5$  (με τις κατάλληλες μονάδες ώστε οι τάσεις και το ρεύμα να είναι σε V και A αντίστοιχως). Έχουμε επομένως για το ρεύμα της ελεγχόμενης πηγής,

$$I = 1E-3 + (7E-3)*V(1,0) + (8E-5)*V(3,4)$$

Γ) Θα χρησιμοποιήσουμε μια μη γραμμική VCCS για να φτιάξουμε ένα μοντέλο μιας μη γραμμικής αντίστασης. Θεωρούμε μια VCCS της οποίας οι κόμβοι ελέγχου συμπίπτουν με τους κόμβους μεταξύ των οποίων ρέει το ρεύμα της πηγής. Τότε το ρεύμα είναι μη γραμμική συνάρτηση της τάσης. Αυτό προφανώς σημαίνει μη γραμμική αντίσταση. Βλέπε Σχήμα 10.

Οι εντολές είναι,

NONLINR.CIR TO MODEL A NONLINEAR RESISTOR

\* A NONLINEAR VCCS IS USED TO MAKE A RESISTOR WITH I-V GRAPH  
\* TO BE A CURVE (NO STRAIGHT LINE).

VSWEEP 5 0 DC

VSENSE 5 8 0

\* IN THE VCCS BELOW (GNLR), THE CURRENT THAT FLOWS BETWEEN

\* NODES 8 AND 0 IS CONTROLLED BY THE VOLTAGE BETWEEN NODES 8

\* AND 0.

GNLR 8 0 8 0 0 1 4


.DC VSWEEP 0 1 0.05

.END

Η GNLR είναι μια VCCS συνδεδεμένη μεταξύ των κόμβων 8 και 0, και κόμβοι ελέγχου είναι οι ίδιοι, δηλαδή 8 και 0. Η σχέση για το ρεύμα μέσα από την πηγή GNLR είναι,

$$I = 1*V + 4*V^2$$

Όπου V είναι η τάση  $V(8,0)$  και I είναι το ρεύμα που ρέει από τον κόμβο 8 προς τον κόμβο 0 μέσα από την GNLR.



Σχήμα 10

Αν προχωρήσει κάποιος στην ανάλυση με το SPICE θα δει ότι η GNLR συμπεριφέρεται ως μη γραμμική αντίσταση εξαρτώμενη από την τάση στα άκρα της.

- **Μη γραμμικές πηγές τάσης ελεγχόμενες από τάση.**

Η εντολή είναι της μορφής,

```
Exxxxxx N+ N- <POLY(ND)> NC1+ NC1- <NC2+ NC2->...  
+ P0 P1 ...<IC=...>
```

Η σημασία των συμβόλων είναι όπως στα προηγούμενα, άρα προφανής.

Παραδείγματα:

A) ESIMPLE 1 4 7 2 0 0.5 0.3 0.06

Η τάση της πηγής ESIMPLE τύπου VCVS (που εμφανίζεται μεταξύ των σημείων 1 και 4 ) είναι συνάρτηση της τάσης του κόμβου 7 ως προς τον κόμβο 2 και ακολουθεί τη σχέση,

$$V(1,4) = 0.5 \cdot V(7,2) + 0.3 \cdot V(7,2)^2 + (6E-2) \cdot V(7,2)^3$$

Δεν υπάρχει POLY(ND) άρα πρόκειται για εξαρτώμενη πηγή από μια μόνο τάση.

B) EPRODUCT 5 7 POLY(2) 20 22 31 33 0 0 0 0 1

Η πηγή VCVS EPRODUCT δημιουργεί μια τάση μεταξύ των κόμβων 5 και 7 (+ και – αντίστοιχα). Η τάση αυτή είναι ίση με το γινόμενο της τάσης του κόμβου 20 ως προς τον 22 και της τάσης του κόμβου 31 ως προς τον 33. Η εξίσωση για την τάση της πηγής είναι,

$$V(5,7) = 1.0 \cdot V(20,22) \cdot V(31,33)$$

Μια τέτοια διάταξη μπορεί να χρησιμοποιηθεί για διαμόρφωση πλάτους, AM (amplitude modulation) . Μπορεί να είναι τύπου διπλής πλευρικής συχνότητας με φέρουσα (DSB-FC, double sideband full carrier) ή DSB-SC (double sideband suppressed carrier) δηλαδή διπλής πλευρικής συχνότητας χωρίς φέρουσα. Αυτό γίνεται με τη μια ελέγχουσα τάση να είναι αρμονική με το χρόνο με τη φέρουσα συχνότητα και την άλλη ελέγχουσα τάση να είναι το σήμα διαμόρφωσης συν μια σταθερά για την περίπτωση DSB-FC ή χωρίς τη σταθερά, για τη περίπτωση DSB-SC .

- **Μη γραμμικές πηγές ρεύματος ελεγχόμενες από ρεύμα.**

Η εντολή είναι,

```
Fxxxxxx N+ N- <POLY(ND)> VN1 <VN2...>
+ P0 P1 ...<IC=...>
```

Για μια τέτοια πηγή CCCS τα σύμβολα έχουν την προφανή σημασία η οποία περιγράφηκε προηγουμένως.

Παράδειγμα

```
FUNKY 9 3 VINPUT 1 0.4 2
```

Η πηγή FUNKY τύπου CCCS ελέγχεται από την ανεξάρτητη πηγή ρεύματος VINPUT

- **Μη γραμμικές πηγές τάσης ελεγχόμενες από ρεύμα.**

Η εντολή είναι,

```
Hxxxxxx N+ N- <POLY(ND)> VN1 <VN2...>
+ P0 P1 ...<IC=...>
```

Η πηγή Hxxxxxx τύπου CCVS είναι συνδεδεμένοι στους κόμβους N+ και N- και το ρεύμα ρέει μέσα στη πηγή από τον N+ προς τον N- . VN1, VN2 ... είναι τα ονόματα των πηγών τάσης μέσα από τις οποίες ρέουν τα ρεύματα ελέγχου και είναι μια για κάθε διάσταση. Η προκαθορισμένη διάσταση είναι 1 . Η αρχική συνθήκη IC= δίνει τη τιμή των ρευμάτων ελέγχου τη χρονική στιγμή μηδέν. Η προκαθορισμένη αρχική τιμή είναι μηδέν.

Παράδειγμα:

```
HITHERE 7 1 VX 0 2 0 4E2
```

Η πηγή τάσης HITHERE συνδέεται μεταξύ των κόμβων 7 και 1 και ελέγχεται από το ρεύμα που διέρχεται μέσα από την ανεξάρτητη πηγή τάσης VX .

Η σχέση για τη τάση στα άκρα της HITHERE είναι,

$$V(7,1)=0 + 2 \cdot I(VX) + 4E2 \cdot (I(VX))^3 \quad (\text{τάση σε V} \quad \text{ρεύμα σε A})$$

Επειδή δεν καθορίστηκε POLY(ND) η HITHERE είναι συνάρτηση ενός μόνο ρεύματος.



## ΜΟΡΦΕΣ ΑΝΑΛΥΣΗΣ

Η ανάλυση που θα γίνει στο κύκλωμα καθορίζεται από εντολές ελέγχου στο αρχείο εισόδου.

- `.DC` Με αυτή την εντολή προσδιορίζεται το σημείο λειτουργίας dc του κυκλώματος. Οι πυκνωτές αποτελούν διακοπές στο κύκλωμα και οι αυτεπαγωγές βραχυκυκλώματα. Μπορεί να γίνεται μεταβολή τάσης ή ρεύματος πηγής συνεχούς κατά βήματα και κάθε φορά προσδιορισμός του σημείου λειτουργίας DC .

Έχουμε την εντολή ελέγχου,

`.DC SRC START STOP INCR <SRC2 START2 STOP2 INCR2>`

SRC είναι η ανεξάρτητη πηγή τάσης ή ρεύματος που μεταβάλλεται.

START είναι η αρχική τιμή τάσης ή ρεύματος , STOP είναι η τελική τιμή, INCR είναι το βήμα της μεταβολής. Οι τιμές είναι σε volt ή ampere

Είναι δυνατόν να οριστεί μια δεύτερη πηγή SRC2 με τις αντίστοιχες παραμέτρους της.

Αν υπάρχει δεύτερη πηγή γίνεται σάρωση των τιμών της πρώτης για κάθε μία τιμή της δεύτερης.

Πρέπει να υπάρχουν εντολές `.PRINT` και `.PLOT` για να γραφτούν τα αποτελέσματα στο αρχείο εξόδου, xxx.out.

- `.AC` Με αυτή την εντολή γίνεται γραμμική ανάλυση μικρών σημάτων. Το SPICE βρίσκει το σημείο λειτουργίας dc και υπολογίζει τιμές για μοντέλα μικρών σημάτων για όλες τις μη γραμμικές διατάξεις (ημιαγωγούς, μη γραμμικές πηγές κτλ). Στη συνέχεια κάνει ανάλυση για κάθε συχνότητα που έχει καθοριστεί. Τα αποτελέσματα μπορεί να τυπωθούν ή να παρασταθούν γραφικά.

`.AC LIN NP FSTART FSTOP`

`.AC DEC ND FSTART FSTOP`

`.AC OCT NO FSTART FSTOP`

FSTART είναι η ελάχιστη συχνότητα (δεν μπορεί να είναι 0 ή αρνητική) και

FSTOP είναι η μέγιστη συχνότητα.

Οι συχνότητες στις οποίες γίνεται ανάλυση μπορεί να καθοριστούν με τρεις τρόπους,

LIN NP σημαίνει ότι θα γίνει ανάλυση σε NP πλήθος συχνοτήτων που είναι ισαπέχουσες (γραμμικά) μεταξύ FSTART και FSTOP. Στη περίπτωση μιας μόνο συχνότητας θέτουμε NP=1 και FSTART = FSTOP = η επιθυμητή συχνότητα.

DC ND σημαίνει ότι η περιοχή συχνοτήτων υποδιαιρείται σε δεκάδες με ND συχνότητες ανά δεκάδα. Οι συχνότητες κατανομούνται λογαριθμικά. Αν FSTOP δεν είναι κάποιος ακέραιος αριθμός (πλήθος) δεκάδων πάνω από το FSTART, τότε η μέγιστη συχνότητα μπορεί να υπερβεί το FSTOP.

OCT NO σημαίνει ότι το εύρος συχνοτήτων χωρίζεται σε οκτάβες με NO συχνότητες ανά οκτάβα. Οι συχνότητες μεταβάλλονται λογαριθμικά. Αν η FSTOP

δεν είναι ακέραιο πολλαπλάσιο οκτάβων πάνω στο FSTART, η μέγιστη συχνότητα μπορεί να υπερβεί την FSTOP.

Σε αντίθεση με την περίπτωση της εντολής ελέγχου .DC εδώ δεν καθορίζεται ποιας πηγής η συχνότητα θα μεταβληθεί. Όλες οι πηγές ac μεταβάλλονται συγχρόνως. Χρειάζεται προσθήκη των εντολών όπως .PRINT και .PLOT για να γραφτεί η πληροφορία στο xxx.out.

- .TRAN Αυτή η εντολή ελέγχου σημαίνει transient (μεταβατική, εξαρτώμενη από το χρόνο) ανάλυση. Γίνεται ανάλυση όπου οι μεταβλητές υπολογίζονται συναρτήσει του χρόνου.

.TRAN TSTEP TSTOP <TSTART <TMAX>> <UIC>

TSTEP είναι το βήμα χρόνου για εκτύπωση και γραφήματα. TSTOP είναι ο μέγιστος (τελικός) χρόνος ανάλυσης. TSTART είναι το σημείο από όπου αρχίζει η εκτύπωση και το γράφημα. Υπάρχει προκαθορισμένη τιμή που είναι μηδέν. Η ανάλυση γίνεται πάντα από χρόνο μηδέν αλλά τα αποτελέσματα δεν παρουσιάζονται στο xxx.out. TMAX είναι το μέγιστο βήμα για τους υπολογισμούς με το SPICE. Αν δεν καθοριστεί θα ληφθεί από το SPICE το ελάχιστο μεταξύ των TSTEP και (TSTEP-TSTART)/50. Μπορεί να χρησιμοποιηθεί το TMAX για σιγουριά ότι το βήμα για τους υπολογισμούς είναι μικρότερο από το TSTEP παρουσίασης των αποτελεσμάτων. UIC σημαίνει χρήση των αρχικών συνθηκών. Αν υπάρχει το UIC τότε το SPICE δεν υπολογίζει το σημείο ηρεμίας πριν την μεταβατική ανάλυση. Τότε το SPICE κάνει το ένα από τα ακόλουθα δύο πράγματα:

1. Αν δεν υπάρχει η εντολή .IC χρησιμοποιεί τις αρχικές συνθήκες που δίνονται (για πυκνωτές, αυτεπαγωγές και ημιαγωγούς) ως το αρχικό σημείο για την ανάλυση συναρτήσει του χρόνου.
  2. Αν υπάρχει η εντολή .IC θα χρησιμοποιήσει τις τάσεις των κόμβων στην εντολή .IC για να προσδιορίσει τις αρχικές συνθήκες για τα διάφορα στοιχεία στο κύκλωμα.
- .OP (operating point) κάνει το SPICE να υπολογίσει το σημείο dc και να τυπώσει τα αποτελέσματα. Ο υπολογισμός γίνεται αυτόματα αν γίνεται ανάλυση AC ή όταν γίνεται μεταβατική ανάλυση, όμως τότε δεν τυπώνονται τα αποτελέσματα.
  - .TF Με αυτή την εντολή ελέγχου γίνεται ανάλυση τύπου Συνάρτησης Μεταφοράς (Transfer Function) για μικρά σήματα.

.TF OUTPUTVAR INPUTSRC

OUTPUTVAR είναι η μεταβλητή εξόδου μικρού σήματος (τάση ή ρεύμα).

INPUTSRC είναι η μεταβλητή εισόδου μικρού σήματος (τάση ή ρεύμα).

Η εντολή ελέγχου .TF κάνει ανάλυση συνάρτησης μεταφοράς μικρού σήματος dc και τυπώνει στο αρχείο εξόδου την τιμή της αντίστασης εισόδου (μέγεθος

INPUTSCR) και της αντίστασης εξόδου (μέγεθος OUTPUTVAR) και επίσης την τιμή κάποιου είδους συνάρτησης μεταφορά όπως, ενίσχυση τάσης, ενίσχυση ρεύματος, διαντίσταση ή διαγωγιμότητα.

Παραδείγματα:

.TF V(5) VIN

Εδώ έξοδος θα είναι η τάση στον κόμβο 5 και είσοδος θα είναι η πηγή τάσης που ορίζεται ως VIN. Η συνάρτηση μεταφοράς θα είναι η ενίσχυση τάσης.

.TF I(VIDRAIN) VGAIN

Έξοδος θα είναι το ρεύμα δια της πηγής τάσης VIDRAIN ενώ είσοδος είναι η πηγή τάσης VGATE. Η συνάρτηση μεταφοράς είναι η διαγωγιμότητα.

- .SENS Γίνεται ανάλυση ευαισθησίας και βρίσκεται η ευαισθησία σε μικρά σήματα dc για μία ή περισσότερες μεταβλητές εξόδου ως προς κάθε παράμετρο του κυκλώματος.

.SENS OV1 <OV2>...

OV1 είναι η μεταβλητή εξόδου, όπως για παράδειγμα τάση κόμβων ή ρεύμα που διέρχεται μέσα από πηγή τάσης.

- .DISTO Αυτή η εντολή χρησιμοποιείται μαζί με την εντολή .AC για να γίνει ανάλυση παραμόρφωσης .

.DISTO RLOAD INTER <SKW2 <REFPWR <SPW2>>>

Το SPICE θα προσδιορίσει την παραμόρφωση μικρών σημάτων μαζί με την ανάλυση μονίμων φαινομένων ac για μικρά σήματα.

RLOAD είναι το όνομα της αντίστασης εξόδου μέσα στην οποία θα υπολογιστεί όλη η ισχύς παραμόρφωσης. INTER είναι το διάστημα στο οποίο θα τυπωθεί μία περίληψη της συμβολής όλων των μη γραμμικών διατάξεων στην ολική παραμόρφωση.

Αν δεν δοθεί το INTER ή του δοθεί τιμή μηδέν, τότε δεν γίνεται εκτύπωση περίληψης. Για παράδειγμα, αν INTER=2 θα γίνει περίληψη κάθε δεύτερη συχνότητα ανάλυσης ac. Η ανάλυση παραμόρφωσης γίνεται με μια ή δύο συχνότητες στην είσοδο. Έστω  $f_1$  η συχνότητα της ανάλυσης ac, τότε η  $f_2 = SKW2 * f_1$ .

Αν η SKW2 παραληφθεί τότε  $f_2 = 0,9 * f_1$  (δηλαδή προκαθορισμένη τιμή για το SKW2 είναι το 0,9).

REFPWR είναι η στάθμη ισχύος αναφοράς που χρησιμοποιείται για τον υπολογισμό της παραμόρφωσης. Αν δεν δοθεί τότε το REFPWR είναι 1mW

(0 dBmW). SPW2 είναι το πλάτος για τη συχνότητα  $f_2$ , αν παραληφθεί είναι  $SPW2=1$ .

Η έξοδος που προκύπτει από την εντολή .DSTO περιέχει τα εξής,

HD2, το πλάτος της δεύτερης αρμονικής της  $f_1$ , υποθέτοντας ότι δεν υπάρχει η  $f_2$ .

HD3, το πλάτος της τρίτης αρμονικής της  $f_1$  χωρίς την ύπαρξη της  $f_2$ .

SIM2, το πλάτος της συνιστώσας ( $f_1 + f_2$ ).

DIM2, το πλάτος της συνιστώσας ( $f_1 - f_2$ ).

DIM3, το πλάτος της συνιστώσας ( $2f_1 - f_2$ ).

- .NOISE Μαζί με την εντολή .AC υπολογίζεται ο ισοδύναμος θόρυβος εξόδου και εισόδου σε καθορισμένα σημεία εισόδου και εξόδου.

#### .NOISE OUTPUTV INPUTSRC NUMSUM

Με αυτή την εντολή γίνεται ανάλυση θορύβου μαζί με την ανάλυση ac μικρού σήματος μονίμου φαινομένου. OUTPUTV είναι μια τάση η οποία είναι το σημείο άθροισης εξόδου για το θόρυβο. INPUTSRC είναι το όνομα του μιας ανεξάρτητης πηγής τάσης ή ρεύματος που θα είναι η είσοδος θορύβου αναφοράς.

NUMSUM είναι το διάστημα περίληψης, δηλαδή το διάστημα για το οποίο θα τυπωθεί περίληψη των συμβολών όλων των γεννητριών θορύβου.

Αν δε δοθεί το NUMSUM ή γίνει μηδέν δεν δίνεται περίληψη.

Αν  $NSUM = 5$  (για παράδειγμα) τότε θα έχουμε περίληψη κάθε  $5^{\text{η}}$  συχνότητα ανάλυσης ac. Προσοχή μπορεί να ξοδευτεί πολύ χαρτί για το τύπωμα!!

#### Παράδειγμα

.NOISE V(2,3) VIN1 10

Αυτή η εντολή σημαίνει ότι θα γίνει ανάλυση μαζί με την ανάλυση ac. Θα γίνει περίληψη για εκτύπωση κάθε  $10^{\text{η}}$  συχνότητα ac. Ο θόρυβος εξόδου θα μετριέται ως η διαφορά τάσης μεταξύ των κόμβων 2 και 3 και ο ισοδύναμος θόρυβος εισόδου θα αναφέρεται στην πηγή τάσης VIN1.

- .FOUR Γίνεται ανάλυση Fourier μιας μεταβλητής εξόδου. Πρέπει να γίνεται μαζί με μεταβατική ανάλυση. Υπολογίζονται τα πλάτη και οι σχέσεις των πρώτων 9 αρμονικών μιας οριζόμενης θεμελιώδους καθώς και η συνιστώσα dc.

.FOUR FREQ OV1 <OV2 OV3...>

Πρέπει να υπάρχει .TRAN! Γίνεται προσδιορισμός των πλατών της dc και των πρώτων 9 συχνοτήτων μεταξύ των οποίων είναι και η θεμελιώδης.

FREQ είναι η θεμελιώδης συχνότητα. OV1 είναι η μεταβλητή (εξόδου) στην οποία γίνεται η ανάλυση Fourier. Η ανάλυση δεν γίνεται σε όλο το χρονικό διάστημα του

μεταβατικού φαινομένου αλλά στο χρονικό διάστημα από ( TSTART – (χρόνος μιας περιόδου)) μέχρι TSTOP. Γι' αυτό πρέπει το χρονικό διάστημα της μεταβατικής ανάλυσης να είναι τουλάχιστον 1/FREQ.

Για να υπάρχει ικανοποιητική ακρίβεια πρέπει το μέγιστο βήμα χρόνου υπολογισμού (TMAX στο .TRAN) να τεθεί ίσο με (χρόνος μιας περιόδου) /100 ή μικρότερο.

Παράδειγμα

```
.FOUR 50 V(6) I(VSOURCE)
```

Σύμφωνα με αυτή την εντολή, θα γίνει ανάλυση Fourier για την τάση στον κόμβο 6 και επίσης για το ρεύμα δια μέσου της πηγής τάσης VSOURCE. Η θεμελιώδης συχνότητα είναι 50Hz.

- .TEMP Αυτή η εντολή λέει στο SPICE σε τι θερμοκρασία να κάνει ανάλυση του κυκλώματος.

```
.TEMP T1 <T2<T3...>>
```

Τα T1,T2,T3... είναι σε βαθμούς Κελσίου. Η ελάχιστη θερμοκρασία είναι –223,0 °C. Χαμηλότερες θερμοκρασίες αγνοούνται. Η ονομαστική θερμοκρασία είναι TNOM=27 °C, εκτός αν χρησιμοποιηθεί στην εντολή .OPTIONS άλλη τιμή. Οι παράμετροι στα μοντέλα ημιαγωγών και αντιστατών που μεταβάλλονται με τη θερμοκρασία δίνονται στη θερμοκρασία TNOM.

Παράδειγμα:

```
.TEMP 0 25 60 100
```

Η ανάλυση θα γίνει σε θερμοκρασία 0 °C, 25 °C, 60 °C και 100 °C.

- .IC Αυτή η εντολή μπορεί και καθορίζει αρχικές συνθήκες για την μεταβατική ανάλυση.

```
.IC V(NODENUM1)= VAL1 V(NODENUM2)= VAL2...
```

NODENUM1 είναι κόμβος (≠0) και VAL1 είναι η τιμή της τάσης του κόμβου. Ανάλογα ισχύουν για NODENUM2 κτλ.

Υπάρχουν δύο ερμηνείες της ανωτέρω εντολής που εξαρτώνται από την παράμετρο UIC στο .TRAN.

- 1) Αν υπάρχει η UIC στο .TRAN, τότε στον υπολογισμό των αρχικών συνθηκών για πυκνωτές, BJT, JFET και MOSFET χρησιμοποιούνται οι τιμές τάσεων των κόμβων που καθορίστηκαν στο .IC. Αυτό ισοδυναμεί με καθορισμό της παραμέτρου IC = ... για κάθε μία διάταξη αλλά είναι πιο βολικό. Αν καθοριστεί η

παράμετρος IC = ... μίας διάταξης αυτό υπερτερεί των τιμών που καθορίζει η εντολή .IC για την ίδια διάταξη. Δεν γίνεται ανάλυση dc πριν την μεταβατική ανάλυση. Γιαυτό πρέπει να καθοριστούν οι τάσεις dc των πηγών με χρήση της

- 2) εντολής .IC αν αυτό χρειάζεται για να υπολογιστούν αρχικές συνθήκες της διάταξης.
- 3) Αν δεν υπάρχει το UIC στην εντολή .TRAN, τότε θα γίνει ανάλυση πόλωσης dc (αρχική μεταβατική) πριν την μεταβατική ανάλυση. Οι τάσεις των κόμβων που καθορίζονται στην εντολή .IC θα αναγκαστούν να πάρουν τις επιθυμητές αρχικές τιμές πόλωσης, κατά την ανάλυση πόλωσης. Κατά την μεταβατική ανάλυση, δεν θα ληφθεί υπόψη αυτός ο περιορισμός για τους κόμβους.

- .WIDTH: Η εντολή είναι,

WIDTH IN = COLNUMIN OUT = COLNUMOUT

Όπου COLNUMIN είναι ο αριθμός της τελευταίας στήλης του αρχείου εισόδου που διαβάζει το SPICE. Η εντολή εκτελείται αμέσως μετά την εισαγωγή της. Αυτό σημαίνει ότι πρέπει να γράφεται μετά τον τίτλο του αρχείου εισόδου. Το εύρος του αρχείου εξόδου είναι COLNUMOUT, οι επιτρεπτές τιμές είναι 80 και 133. Καλό είναι για να μην υπάρχουν προβλήματα με μερικούς υπολογιστές που έχουν σαν προκαθορισμένη τιμή τις 133 στήλες, κατά την παρουσίαση των αποτελεσμάτων στην οθόνη, να μπαίνει η εντολή ελέγχου .WIDTH OUT = 80 που περιορίζει το εύρος στους 80 χαρακτήρες.

- .OPTIONS: Υπάρχουν 32 δυνατότητες (options) που μπορούν να ορισθούν. Μεταξύ αυτών αναφέρουμε ενδεικτικά τη δυνατότητα για μεταβολή της προκαθορισμένης ονομαστικής θερμοκρασίας. TNOM=x μεταβάλλει την ονομαστική θερμοκρασία. Η προκαθορισμένη τιμή της είναι 27 °C.

.OPTIONS OPT1 OPT2 ή και OPT=OPTVAL...

Παράδειγμα

.OPTIONS TNOM=0

Η εντολή αυτή κάνει την ονομαστική θερμοκρασία 0 °C.

- .NODESET : Η εντολή είναι,

.NODESET V(NODENUM1) = VAL1 V(NODENUM2) = VAL2...

Όπου NODENUM είναι θετικός ακέραιος που δηλώνει κόμβο (όχι τον κόμβο 0) και VAL είναι η τάση αυτού του κόμβου. Το SPICE θα κάνει την αρχική προσπάθεια να βρει λύση για ένα σημείο πόλωσης DC ή μεταβατικό, με τους κόμβους που ορίστηκαν κρατημένους στις τάσεις που δόθηκαν. Δεν χρειάζεται να δοθούν οι τάσεις σε όλους τους κόμβους. Κατόπιν αυτός ο περιορισμός αίρεται και το SPICE συνεχίζει

προς τη σωστή λύση. Αυτή η εντολή μπορεί να είναι χρήσιμη σε ασταθή ή δισταθή κυκλώματα, όπου ο προγραμματιστής μπορεί να εικάσει τιμές για μερικούς κόμβους.

## ΕΝΤΟΛΕΣ ΕΞΟΔΟΥ

**.OP** : Δεν υπάρχουν παράμετροι που πρέπει να προσδιοριστούν. Αυτή η εντολή κάνει το SPICE να λύσει το πρόβλημα και να βρει το σημείο ηρεμίας (σημείο πόλωσης) του κυκλώματος, με τους πυκνωτές «ανοιχτούς» και τις επαγωγές «βραχυκυκλωμένες», και να τυπώσει τα λεπτομερή αποτελέσματα της ανάλυσης μεγέθι όπως, η καταναλισκόμενη ισχύς στο κύκλωμα και στις πηγές καθώς και οι παράμετροι μικρού σήματος των ημιαγωγών και μη γραμμικών διατάξεων. Το SPICE κάνει τέτοια ανάλυση χωρίς την εντολή **.OP** όταν γίνεται ανάλυση AC ή μεταβατική αλλά τότε δεν τυπώνει τα αποτελέσματα.

**.PRINT** : Η εντολή είναι,

**.PRINT ANALTYPE OV1 <OV2 ... OV8>**

**ANALTYPE** δηλώνει έναν από τους ακόλουθους τύπους ανάλυσης,

**DC, AC, TRAN, DISTO, NOISE** .

**OV** δηλώνει την μεταβλητή εξόδου. Μπορεί να υπάρχουν μαζί μέχρι 8 τύποι ανάλυσης σε μια εντολή **.PRINT** .

**.PLOT** : Η εντολή αυτή κάνει γραφικές παραστάσεις σε εκτυπωτές. Η μορφή της εντολής είναι,

**.PLOT ANALTYPE OV1 <(PLO1,PHI1)> <OV2 <(plo2,PHI2)>  
+ ... OV8 <(plo8,PHI8)>>**

Μπορεί να υπάρχουν μέχρι 8 μεταβλητές **OV** . Η ανεξάρτητη μεταβλητή στο γράφημα («οριζόντιος άξονας») θα είναι: η τάση DC όταν πρόκειται για ανάλυση DC, η συχνότητα όταν πρόκειται για ανάλυση AC ή ο χρόνος για μεταβατική ανάλυση. Τα (PLO, PHI) είναι τα όρια του γραφήματος και είναι προαιρετικά.

**.PROBE** : Αυτή είναι εντολή του PSpice για γραφήματα. Η σύνταξη της εντολής είναι,

**.PROBE <VNAME1,VNAME2, ... >**

**VNAME** είναι προαιρετικά και δηλώνουν τις μεταβλητές για τις οποίες θέλουμε να γίνουν γραφήματα. Αν δεν δώσουμε συγκεκριμένες μεταβλητές προκαθορισμένες τιμές είναι ΟΛΕΣ.

Υπάρχουν διάφοροι τρόποι χρήσης του Probe που μπορεί ο ενδιαφερόμενος να βρει διαβάζοντας το αρχείο Help του PSpice και γενικώς ασκούμενος με το πρόγραμμα. Εμείς θα αναφερθούμε σε έναν σχετικά απλό τρόπο αφετηρία για χρήση του Probe. Αφού τρέξει το πρόγραμμα επιλέγουμε με το ποντίκι **Run Probe**, στη συνέχεια στην

εικόνα για τη γραφική παράσταση επιλέγουμε **Add Trace** και στη συνέχεια τις μεταβλητές για τις οποίες θέλουμε τις γραφικές τους παραστάσεις.

## ΗΜΙΑΓΩΓΟΙ ΣΤΟ SPICE

Το SPICE έχει μοντέλα για ημιαγωγούς.

- Μοντέλο διόδου.

Η εντολή για δίοδο διεπαφής έχει τη μορφή,

Dxxxxxx N+ N- MODNAME <AREA> <OFF> <IC=VD>

Dxxxxxx είναι το όνομα της διόδου. N+ είναι ο κόμβος όπου είναι συνδεδεμένη η «άνοδος» της διόδου και N- είναι ο κόμβος που είναι συνδεδεμένη η «κάθοδος». MODNAME είναι το όνομα του μοντέλου που χρησιμοποιείται στην εντολή ελέγχου .MODEL . Οι προαιρετικές παράμετροι είναι,

AREA - είναι ο συντελεστής επιφάνειας ο οποίος μας λέει πόσες διόδοι του τύπου του μοντέλου MODNAME συνδέονται παράλληλα και φτιάχνουν τη δίοδο Dxxxxxx .

OFF - αρχική συνθήκη του Dxxxxxx για την ανάλυση dc .

IC=VD - κάνει το SPICE να χρησιμοποιήσει το VD ως αρχική συνθήκη για την τάση της διόδου αντί για την τάση ηρεμίας της διόδου, όταν γίνεται μεταβατική ανάλυση.

Η γενική μορφή της εντολής για το μοντέλο διόδου διεπαφής είναι,

.MODEL MODNAME D <(PAR1=PVAL1 PAR2=PVAL2...)>

MODNAME είναι το όνομα του μοντέλου που δώσαμε στη δίοδο σε κάποια εντολή στοιχείου κυκλώματος. Το D δηλώνει ότι ο ημιαγωγός είναι δίοδος. PARxx = PVALxxx δίνει την τιμή στην κάθε παράμετρο από αυτές του Πίνακα 3 .

### ΠΙΝΑΚΑΣ 3

Παράμετροι μοντέλου διόδου.				
#	Σύμβολο	Παράμετρος	Μονάδες	Προκαθορισμένες τιμές
1	IS	saturation current	A	$1,0 \times 10^{-14}$
2	RS	ohmic resistance	$\Omega$	0
3	N	emission coefficient	-	1
4	TT	transit time	s	0
5	CJO	zero bias junction capacitance	F	0
6	VJ	junction potential	V	1
7	M	grading coefficient	-	0,5
8	EG	activation energy	eV	1,11
9	XTI	saturation current temperature exponent	-	3,0
10	KF	flicker noise coefficient	-	0
11	AF	flicker noise exponent	-	1
12	FC	coefficient for forward bias depletion	-	0,5



		capacitance formula		
13	BV	reverse breakdown voltage	V	Infinite
14	IBV	current at breakdown voltage	A	$1,0 \times 10^{-3}$

Παράδειγμα:

D1 7 9 H11 (εντολή στοιχείου διόδου).  
.MODEL H11 D(RS=0.5 BV=400 IBV=50M)

Οι εντολές αυτές δηλώνουν ότι, η διόδος D1 είναι συνδεδεμένη μεταξύ των κόμβων N+ = 7 και N- = 9 και το μοντέλο έχει όνομα H11. Η άνοδος είναι στο 7 και η κάθοδος στο 9. Το μοντέλο H11 λέει ότι η ωμική αντίσταση της διόδου είναι 0,5 Ω, η ανάστροφη τάση διάσπασης είναι 400 V και το ρεύμα στην ανάστροφη τάση διάσπασης είναι 50 mA

### • Μοντέλο διπολικού τρανζίστορ (BJT).

Η εντολή για διπολικό τρανζίστορ είναι,

Qxxxxxx NC NB NE <NS> MODNAME <AREA><OFF><IC=VBE,VCE>

Το Q δηλώνει διπολικό τρανζίστορ. Qxxxxxx είναι το όνομα του τρανζίστορ. NC, NB και NE είναι οι κόμβοι όπου συνδεδεμένα αντίστοιχα ο συλλέκτης, η βάση και ο εκπομπός. MODNAME είναι το όνομα του μοντέλου που χρησιμοποιείται σε μια εντολή ελέγχου που λέγεται .MODEL . Υπάρχουν και προαιρετικές παράμετροι που είναι,

NS - δηλώνει τον κόμβο όπου είναι συνδεδεμένο το υπόστρωμα (substrate), αν δεν δίνεται τότε ισχύει προκαθορισμένη τιμή που είναι ο κόμβος 0.

AREA - είναι ο συντελεστής επιφάνειας ο οποίος καθορίζει πόσα διπολικά τρανζίστορ (BJT) με μοντέλο το MODNAME τοποθετούνται παράλληλα για να φτιαχτεί ένα Qxxxxxx .

OFF - αρχική συνθήκη για το Qxxxxxx για ανάλυση dc .

IC = VBE, VCE - καθορισμός αρχικών συνθηκών, για χρήση με την δυνατότητα UIIC της εντολής ελέγχου .TRAN . Αυτό δεσμεύει το SPICE να χρησιμοποιήσει ως αρχικές συνθήκες τα VBE , VCE αντίστοιχως για τις τάσεις βάσης-εκπομπού και συλλέκτη-εκπομπού, αντί για τις τάσεις διεπαφής του σημείου ηρεμίας όταν γίνεται μεταβατική ανάλυση.

Η γενική εντολή για το μοντέλο διπολικού τρανζίστορ είναι,

.MODEL MODNAME NPN <(PAR1=PVAL1 PAR2=PVAL2 ...)>

Αντί για NPN μπορεί να έχουμε PNP. Υπάρχουν 40 παράμετροι (PAR με τιμές PVAL) για το τρανζίστορ. Βλέπε Πίνακα 4 .

## ΠΙΝΑΚΑΣ 4

### ( Modified Gummel-Poon BJT Παράμετροι)

Παράμετροι μοντέλου διπολικού τρανζίστορ				
#	Όνομα	Παράμετρος	Μονάδες	Προκαθορισμένες τιμές
1	IS	transport saturation current	A	$1,0 \times 10^{-16}$
2	BF	ideal maximum forward beta	-	100
3	NF	forward current emission coefficient	-	1,0
4	VA	forward early voltage	V	Infinite
5	IKF	corner for forward beta high current roll-off	A	Infinite
6	ISE	B-E leakage saturation current	A	$1,0 \times 10^{-13}$
7	NE	B-E leakage emission coefficient	-	1,5
8	BR	Ideal maximum reverse beta	-	1
9	NR	reverse current emission coefficient	-	1
10	VAR	reverse Early voltage	V	Infinite
11	IKR	corner for reverse beta high current roll-off	A	Infinite
12	ISC	B-C leakage saturation current	A	0
13	NC	B-C leakage emission coefficient	-	21,5
14	RB	zero bias base resistance	Ohms	0
15	IRB	current where base resistance falls halfway to its min value	A	Infinite
16	RBM	minimum base resistance at high currents	Ohms	RB
17	RE	emitter resistance	Ohms	0
18	RC	collector resistance	Ohms	0
19	CJE	B-E zero-bias depletion capacitance	F	0
20	VJE	B-E built-in potential	V	0,75
21	MJE	B-E junction exponential factor	-	0,33
22	TF	ideal forward transit time	sec	0
23	XTF	coefficient for bias dependence of TF	-	0
24	VTF	voltage describing VBC dependence of TF	V	Infinite
25	IRF	high-current parameter for effect of TF	A	0
26	PRF	excess phase at freq=1,0/(TF*2PI)Hz	deg	0
27	CJC	B-C zero-bias depletion capacitance	F	0
28	VJC	B-C built-in potential	V	0,75
29	MJC	B-C junction exponential factor	-	0,33
30	XCJC	fraction of B-C depletion capacitance connected to interval base node	-	1
31	TR	ideal reverse transit time	Sec	0
32	CJS	zero-bias collector-substrate capacitance	F	0
33	VJS	substrate junction built-in potential	V	0,75
34	MJS	substrate junction exponential factor	-	0
35	XTB	forward and reverse beta temperature exponent	-	0
36	EG	energy gap for temperature effect on IS	EV	1,11
37	XTI	temperature exponent for effect on IS	-	3
38	KF	flicker-noise coefficient	-	0
39	AF	flicker-noise exponent	-	1
40	FC	coefficient for forward-bias depletion capacitance formula	-	0,5

Παράδειγμα:

Q11 4 10 5 SMALLSIH  
.MODEL SMALLSIH NPN (BF=140)

Αυτές οι εντολές δηλώνουν ότι, το τρανζίστορ Q11 είναι διπολικό (BJT), έχει τον συλλέκτη στον κόμβο 4, τη βάση στον κόμβο 10 και τον εκπομπό στον 5 .

Το τρανζίστορ είναι NPN με  $\beta=140$ .

- **Μοντέλο τρανζίστορ διεπαφής επίδρασης πεδίου (JFET).**

Η εντολή για τρανζίστορ τύπου JFET είναι,

Jxxxxxx ND NG NS MODNAME <AREA> <OFF> <IC=VDS,VGS>

Τα σύμβολα είναι αντίστοιχα με την περίπτωση του διπολικού τρανζίστορ. ND σημαίνει κόμβος της καταβόθρας (drain), NG κόμβος της πύλης (gate) και NS κόμβος της πηγής (SOURCE). Είναι ευνόητο τι σημαίνουν VDS και VGS (κατ'αντιστοιχία με το διπολικό τρανζίστορ).

Για το μοντέλο του τρανζίστορ JFET έχουμε,

.MODEL MODNAME NJF<(PAR1 = PVAL1 PAR2 = PVAL2...)>

NJF σημαίνει ότι πρόκειται για ημιαγωγό καναλιού – N . Αν η παράμετρος αυτή γίνει PJF τότε πρόκειται για ημιαγωγό καναλιού – P . Οι παράμετροι, PAR , φαίνονται στον Πίνακα 5 .

## ΠΙΝΑΚΑΣ 5

Παράμετροι μοντέλου τρανζίστορ JFET				
#	Όνομα	Παράμετρος	Μονάδες	Προκαθορισμένες τιμές
1	VTO	threshold voltage	V	- 2.0
2	BETA	transconductance parameter	A/V <sup>2</sup>	1,0×10 <sup>-4</sup>
3	LAMBDA	channel length modulation parameter	V <sup>-1</sup>	0
4	RD	drain ohmic resistance	Ω	0
5	RS	source ohmic resistance	Ω	0
6	CGS	zero bias G-S junction capacitance	F	0
7	CGD	zero bias G-D junction capacitance	F	0
8	PB	gate junction potential	V	1
9	IS	gate junction saturation current	A	1,0×10 <sup>-14</sup>
10	KF	flicker noise coefficient	-	0
11	AF	flicker noise exponent	-	1
12	FC	coefficient for forward bias depletion capacitance formula	-	0,5

Το μοντέλο που χρησιμοποιείται είναι των Shichman και Hodges.

- **Μοντέλο τρανζίστορ MOS επίδρασης πεδίου (MOSFET).**

Η γενική μορφή της εντολής για τρανζίστορ επίδρασης πεδίου, μετάλλου οξειδίου (MOSFET) είναι,

```
Mxxxxxx ND NG NS NB MODNAME <L=VAL> <W=VAL>  
+<AD=VAL> <AS=VAL> <PD=VAL> <PS=VAL> <NRD=VAL>  
+<NRS=VAL> <OFF> <IC=VDS,VGS,VBS>
```

Τα περισσότερα σύμβολα είναι προφανή (βλέπε προηγούμενο μοντέλο, JFET). NB είναι ο κόμβος που είναι συνδεδεμένο η «μάζα», το υπόστρωμα (bulk ή substrate). Οι προαιρετικές παράμετροι είναι,

L μήκος του καναλιού σε μέτρα  
W εύρος του καναλιού σε μέτρα  
AD επιφάνεια διάχυσης καταβόθρας (drain) σε  $m^2$  .  
AS επιφάνεια διάχυσης πηγής σε  $m^2$  .

Αν δεν δίνονται τιμές για κάποιες από αυτές τις παραμέτρους τότε το πρόγραμμα χρησιμοποιεί τις προκαθορισμένες τιμές. Οι προκαθορισμένες τιμές μπορεί να καθοριστούν και με την εντολή ελέγχου .OPTIONS . Ισχύει η αντιστοιχία DEF L, DEF W, DEF AD, DEF AS . Οι υπόλοιπες προαιρετικές παράμετροι είναι, PD περίμετρος της διεπαφής καταβόθρας σε μέτρα  
PS περίμετρος της διεπαφής πηγής σε μέτρα  
NRD ισοδύναμος αριθμός τετραγώνων της διάχυσης καταβόθρας  
NRS ισοδύναμος αριθμός τετραγώνων της διάχυσης πηγής.

Οι τιμές των NRD και NRS πολλαπλασιάζουν την «επιφανειακή» αντίσταση RSH που καθορίζεται στην εντολή .MODEL με σκοπό την ακριβή εκπροσώπηση της παρασιτικής αντίστασης καταβόθρας και πηγής σε σειρά για το καθένα τρανζίστορ. Οι προκαθορισμένες τιμές είναι, PD+PS+0 , NRD=NRS=1 .

OFF δηλώνει αρχική συνθήκη για ανάλυση dc .

IC βάζει τις αρχικές συνθήκες για τις τάσεις καταβόθρας-πηγής, πύλης-πηγής και υπόβαθρου-πηγής. Αυτό χρησιμοποιείται σε συνδυασμό με την δυνατότητα UIC της εντολής ελέγχου .TRAN .Αυτό αναγκάζει το SPICE να χρησιμοποιήσει τις τιμές των VDS, VGS και VBS αντί για τις τιμές τάσεων του σημείου ηρεμίας όταν γίνεται μεταβατική ανάλυση.

Η εντολή για το μοντέλο MOSFET είναι,

```
.MODEL MODNAME NMOS<(PAR1=PVAL1 PAR@=PVAL2...)>
```

NMOS σημαίνει ότι ο ημιαγωγός είναι τύπου καναλιού-N . Αν η παράμετρος είναι PMOS τότε έχουμε κανάλι – P . PAR είναι το όνομα παραμέτρου όπως φαίνεται στον Πίνακα 6 και PVAL η τιμή της.

Παράδειγμα:

MBUFFER 8 5 7 0 SIMPLE  
.MODEL SIMPLE NMOS

Μπορεί να υπάρχουν πληροφορίες «γεωμετρίας» για το τρανζίστορ που δίνονται με εντολή .OPTIONS ή απλώς λαμβάνονται υπόψη οι προκαθορισμένες τιμές που είναι, DEFL=DEFW=100  $\mu\text{m}$  και DEFAS = 0  $\text{m}^2$ .

## ΠΙΝΑΚΑΣ 6

Παράμετροι τρανζίστορ MOSFET			
#	Όνομα	Παράμετρος	Μονάδες
1	LEVEL	Model index	-
2	VTO	Zero bias threshold voltage	V
3	KP	Transconductance parameter	$\text{A V}^{-2}$
4	GAMMA	Bulk threshold parameter	V
5	PHI	Surface potential	V
6	LAMBDA	Channel length modulation (MOS1 and MOS2 only)	$\text{V}^{-1}$
7	RD	Drain ohmic resistance	$\Omega$
8	RS	Source ohmic resistance	$\Omega$
9	CBD	Zero bias B-D junction capacitance	F
10	CBS	Zero bias B-S junction capacitance	F
11	IS	Bulk junction saturation current	A
12	PB	Bulk junction potential	V
13	CGSO	Gate source overlap capacitance per meter channel width	F/m
14	CGDO	Gate drain overlap capacitance per meter channel width	F/m
15	CGBO	Gate bulk overlap capacitance per meter channel length	F/m
16	RSH	Drain and source diffusion Sheet resistance	$\Omega/\text{square}$
17	CJ	Zero bias bulk junction bottom capacitance per square meter of junction area	$\text{F m}^{-2}$
18	MJ	Bulk junction bottom gradient coefficient	-
19	CJSW	Zero bias bulk junction sidewall capacitance per meter of junction perimeter	F/m
20	MJSW	Bulk junction sidewall grading coefficient	-
21	JS	Bulk junction saturation current per square meter of junction area	$\text{A m}^{-2}$
22	TOX	Oxide thickness	m
23	NSUB	Substrate doping	$\text{cm}^{-3}$
24	NSS	Surface state density	$\text{cm}^{-2}$
25	NFS	Fast surface state density	$\text{cm}^{-2}$
26	TPG	Type of gate material: +1 opposite to substrate -1 same as substrate 0 Al gate	
27	XJ	Metallurgical junction depth	m
28	LD	Lateral diffusion	m
29	UO	Surface mobility	$\text{cm}^2/\text{Vs}$
30	UCRIT	Critical field for mobility degradation (MOS2 only)	V/cm
31	UEXP	Critical field exponent in	-

		Mobility degradation (MOS2 only)		
32	UTRA	Transverse field coefficient (mobility) (deleted for MOS2)	-	0
33	VMAX	Maximum drift velocity of carriers	M/s	0
34	NEFF	Total channel charge (fixed and Mobile) coefficient (MOS2 only)	-	1
35	XQC	Thin oxide capacitance model flag And coefficient of channel charge Share attributed to drain (0-0,5)	-	1
36	KF	Flicker noise coefficient	-	0
37	AF	Flicker noise exponent	-	1
38	FC	Coefficient for forward bias Depletion capacitance formula	-	0,5
39	DELTA	Width effect on threshold voltage (MOS2 and MOS3)	-	0
40	THETA	Mobility modulation (MOS3 only)	V <sup>-1</sup>	0
41	ETA	Static feedback (MOS3 only)	-	0
42	KAPPA	Saturation field factor (MOS3 only)	-	0,2

Το μοντέλο που χρησιμοποιείται για τη χαρακτηριστική I-V του MOSFET καθορίζεται από τη μεταβλητή LEVEL .

LEVEL = 1 σημαίνει μοντέλο Shichman-Hodges

LEVEL = 2 MOS2

LEVEL = 3 MOS3

## ΥΠΟΚΥΚΛΩΜΑΤΑ

Αν ένα τμήμα του κυκλώματος χρησιμοποιείται πολλές φορές (π.χ. κύκλωμα τελεστικού ενισχυτή) μπορεί να είναι καλό να γίνει ένα υποκύκλωμα (κάτι ανάλογο με τις υπορουτίνες στον προγραμματισμό).

Η πρώτη εντολή ενός υποκυκλώματος είναι:

.SUBCKT SUBNAME N1 <N2 N3...>

και η τελευταία εντολή είναι:

.ENDS <SUBNAME>


SUBNAME είναι το όνομα του υποκυκλώματος. N1, N2, N3... είναι εκείνοι οι κόμβοι του υποκυκλώματος που θα συνδεθούν με άλλους κόμβους του κυκλώματος. Αυτοί λέγονται εξωτερικοί κόμβοι του υποκυκλώματος. Οι άλλοι κόμβοι λέγονται εσωτερικοί και ενώ μπορεί να έχουν τους ίδιους αριθμούς με του άλλου κυκλώματος δεν έχουν καμία σχέση με αυτούς εκτός από τον κόμβο 0 που είναι ο ίδιος παντού. Ο κόμβος 0 δεν μπορεί να περιλαμβάνεται στους N1, N2, ... Δεν επιτρέπονται εντολές ελέγχου μέσα στο υποκύκλωμα.

Η εντολή χρήσης των υποκυκλωμάτων είναι Xyyyyyy N1 <N2 N3...> SUBNAME Xyyyyyy είναι το όνομα των στοιχείων κυκλώματος στο «κυρίως» κύκλωμα που χρησιμοποιεί το υποκύκλωμα SUBNAME.

N1, N2,... είναι οι κόμβοι στους οποίους συνδέεται το υποκύκλωμα.

Παράδειγμα:

Βλέπε Σχήμα 11.




Σχήμα 11. RC.CIR

RC.CIR

```

VGEN 10 0 AC 1
.SUBCKT TEE 1 2
RLEFT 1 8 60
RMID 8 0 50
RRIGHT 8 2 80
CRIGHT 2 0 1V
.ENDS TEE
XA 10 20 TEE
XB 20 30 TEE
XC 30 40 TEE
.AC DEC 5 100 1MEG
.END
    
```

Το υποκύκλωμα T φαίνεται στο Σχήμα 12 .



Σχήμα 12. ΥΠΟΚΥΚΛΩΜΑ TEE

```

.SUBCKT TEE 1 2
RLEFT 1 8 60
RMID 8 0 50
RRIGHT 8 2 80
CRIGHT 2 0 1V
.ENDS TEE
    
```

## ΜΕΤΑΒΟΛΗ ΜΟΝΤΕΛΩΝ ΗΜΙΑΓΩΓΩΝ


Πολλές φορές χρειάζεται να μεταβάλλουμε τις παραμέτρους μοντέλων όπως των τρανζίστορ ή διόδων κτλ για να προσεγγίσουμε συγκεκριμένες διατάξεις καλύτερα.

- Μοντέλο dc διόδου

Ισχύει:

$$I_D = I_S [\exp(V_D/N \cdot V_T) - 1]$$

$I_S$  είναι το ρεύμα κόρου,  $N$  είναι ο γνωστός συντελεστής που έχει γενικώς την τιμή 1,0 και  $V_T = kT/e$  ( $k$  = σταθερά Boltzmann =  $1,38 \times 10^{-23}$  J/K,  $T$  η θερμοδυναμική θερμοκρασία και  $e = 1,6 \times 10^{-19}$  C, το θεμελιώδες φορτίο). Στους 25 °C είναι  $kT/e = V_T = 26$  mV. Το προκαθορισμένο  $I_S = 1 \times 10^{-14}$  A.



Σχήμα 13

Το Σχήμα 13 δείχνει το μοντέλο της διόδου. Μπορούμε να φτιάξουμε μοντέλο διόδου με διαφορετικές τιμές παραμέτρων από τις προκαθορισμένες του Πίνακα 3.

```
.MODEL NEWD1 D( IS=5E-9)
```

Αυτό αλλάζει το  $I_S$  από την default τιμή στην τιμή  $I_S = 5 \times 10^{-9}$  A.

- Μεταβολή του μοντέλου του BJT.

Έχουμε, για παράδειγμα, μεταβολή του  $\beta$  του τρανζίστορ με τις εντολές,

```
Q1 9 8 7 NEWT1  
.MODEL NEWT1 NPN(BF=200).
```

Η προκαθορισμένη τιμή του  $\beta$  είναι 100.

Για τις σχετικές παραμέτρους και προκαθορισμένες τιμές τους βλέπε Πίνακα 3



- **Μεταβολή του μοντέλου του JFET.**

Έχουμε, για παράδειγμα, μεταβολή του BETA (διαγωγιμότητα) του τρανζίστορ JFET με τις εντολές,

```
J1 9 8 7 B1  
.MODEL B1 NJF(BETA = 5E-4)
```

Η προκαθορισμένη τιμή του BETA είναι  $1,0 \times 10^{-4} \text{ A/V}^2$ .

Για τις σχετικές παραμέτρους και προκαθορισμένες τιμές τους βλέπε Πίνακα 5.

- **Μεταβολή μοντέλου τρανζίστορ MOSFET.**

Για παράδειγμα θεωρούμε τη μεταβολή της παραμέτρου της διαγωγιμότητας. Βλέπε Πίνακα 6.

```
M1 9 8 7 6 B1  
.MODEL B1 PMOS(KP=3E-5)
```

Η προκαθορισμένη τιμή για την διαγωγιμότητα είναι  $KP=2 \times 10^{-5} \text{ A/V}^2$ .

## ΓΡΑΜΜΕΣ ΜΕΤΑΦΟΡΑΣ

Το SPICE χειρίζεται γραμμές μεταφοράς χωρίς απώλειες. Υπάρχουν δυο μορφές εντολών για γραμμές μεταφοράς,

```
Txxxxxx NA+ NA- NB+ NB- Z0=ZVAL F=FREQ  
+<NL=NLENGTH> <IC=VA,IA,VB,IB>  
ή  
Tyyyyyy NA+ NA- NB+ NB- Z0=ZVAL TD=TVALUE  
+ <IC=VA,IA,VB,IB>
```

NA+ , NA- είναι ο θετικός και αρνητικός κόμβος (πρόσημο δυναμικών) της πλευράς A της γραμμής μεταφοράς. NB+ και NB- είναι οι αντίστοιχοι κόμβοι της πλευράς B. Z0 είναι η χαρακτηριστική εμπέδηση (ωμική αντίσταση) της γραμμής.


FREQ είναι η συχνότητα για την οποία η γραμμή έχει μήκος ίσο με NLENGTH φορές το μήκος κύματος. Το μήκος κύματος είναι το μήκος κύματος για το κενό πολλαπλασιασμένο επί τον συντελεστή ταχύτητας της γραμμής (είναι ο λόγος της φασικής ταχύτητας διάδοσης στο υλικό της γραμμής δια της ταχύτητας διάδοσης στο κενό,  $u/c$ ). Αν δεν δοθεί το NL=NLENGTH τότε η προκαθορισμένη τιμή του είναι NL=0,25, δηλαδή το μήκος της γραμμής είναι  $1/4 \times$  (μήκος κύματος).

TD είναι ο χρόνος καθυστέρησης της γραμμής σε δευτερόλεπτα.

Οι προαιρετικές αρχικές συνθήκες είναι τα VA και IA που είναι η τάση και το ρεύμα στο άκρο A της γραμμής με το VA θετικό αν ο κόμβος V+ είναι πιο θετικός από τον V- . Το ρεύμα IA είναι θετικό αν ρέει προς τη γραμμή στον κόμβο V+ . Αντίστοιχα ισχύουν για τη πλευρά B . Για να ληφθούν υπόψη οι αρχικές συνθήκες πρέπει να υπάρχει η εντολή ελέγχου .TRAN με τη δυνατότητα UIC (use initial conditions, χρησιμοποίησε αρχικές συνθήκες).

Παραδείγματα:

- Μοναδικός παλμός σε γραμμή μεταφοράς



Σχήμα 14

```
TLIN1.CIR SINGLE PULSE INTO TRANSMISSION LINE
TO ILLUSTRATE THE TIME DELAY OF A LINE
VIN 1 0 PULSE(0 10 1N 0 0 3N)
RIN 1 2 50
TSHORT 2 0 3 0 Z0=50 TD=2N
RLOAD 3 0 50
.TRAN 0.2N 7N
.END
```

- Εύρος ζώνης συχνοτήτων για την ισχύ γραμμής μεταφοράς balun.


Το balun σημαίνει, ότι έχουμε διάταξη γραμμής μεταφοράς balanced-unbalanced (σύμμετρου-ασύμμετρου) «μετασχηματιστή». Βλέπε Σχήμα 15. Έχουμε τις εντολές,

```
BALUN.CIR TRANSMISSION LINE BALUN, 1 A CURRENT INPUT.
IIN 0 20 AC 1
T1 20 0 30 0 Z0=75 F=100MEG NL=1.0
T2 30 0 40 0 z0=75 F=100MEG NL=0.5
```

```

RL 30 40 300
.AC LIN 41 60MEG 140MEG
.PRINT AC VR(20) VI(20)
.END

```



Σχήμα 15


- Προσαρμογή εμπέδησης με γραμμή μεταφοράς «μετασχηματιστή»  $\frac{1}{4}$  μήκους κύματος.

Βλέπε Σχήμα 16. Οι εντολές είναι:

```

MATCH.CIR 0.25 WAVELENGTH TRANSMISSION LINE MATCH
VS 1 0 AC 141.42
RS 1 2 50
VSENSE 2 10 0
T1 10 0 3 0 Z0=50 F=50MEG NL=2
T2 3 0 4 0 z0=158.11 F=50MEG NL=0.25
* SQRT(500*50)=158.11, 0.25 WAVE MATCH
T3 4 0 5 0 Z0=50 F=50MEG NL=0.25
* T3 SECTION TRANSFORMS 5 OHM RLOAD TO 500 OHM
RLOAD 5 0 5
.AC LIN 41 10MEG 90MEG
.PRINT AC VM(5) (5,25)
.END

```




Σχήμα 16

## ΜΕΤΑΣΧΗΜΑΤΙΣΤΕΣ ΚΑΙ ΔΙΑΚΟΠΤΕΣ

- **Ιδανικός μετασχηματιστής χωρίς αυτεπαγωγές.**

Στο Σχήμα 17 φαίνεται το σχεδιάγραμμα της διάταξης του ιδανικού μετασχηματιστή δηλαδή μετασχηματιστή χωρίς απώλειες.



Σχήμα 17

Χρησιμοποιούνται δυο ελεγχόμενες πηγές. Η πρώτη είναι πηγή τάσης ελεγχόμενη από τάση (VCVS), EPRI-SEC. Αυτή κάνει την τάση του δευτερεύοντος ίση με το γινόμενο της τάσης του πρωτεύοντος επί το  $N=1/(\text{λόγος μετασχηματιστή})$ . Ο λόγος μετασχηματιστή ορίζεται ως το πηλίκο ( # σπειρών πρωτεύοντος ) / ( # σπειρών δευτερεύοντος ). Η δεύτερη ελεγχόμενη πηγή είναι η FSEC-PRI, που είναι πηγή ρεύματος ελεγχόμενη από ρεύμα, (CCCS). Αυτή κάνει το ρεύμα του πρωτεύοντος ίσο με το ρεύμα του δευτερεύοντος πολλαπλασιασμένο επί  $N$ . Για τον έλεγχο της πηγής FSEC-PRI απαιτείται η γνώση (μέτρηση) της τιμής του ρεύματος του δευτερεύοντος. Αυτό επιτυγχάνεται με χρήση της νεκρής πηγής V-ISEC, η οποία λειτουργεί ως αμπερόμετρο που μετρά το ρεύμα που διέρχεται από την EPRI-SEC.

Οι παρακάτω εντολές περιγράφουν ιδανικό μετασχηματιστή με  $N=2$ . Πρόκειται για μετασχηματιστή ανύψωσης (τάσης) αν οι ακροδέκτες του πρωτεύοντος είναι στους κόμβους 10 και 11.

```
EPRI-SEC 20 13 10 11 2
FSEC-PRI 10 11 V-ISEC 2
V-ISEC 20 12 0
```

Προσοχή αυτό είναι ένα μοντέλο που στηρίζεται στη συμπεριφορά της διάταξης και όχι στη «Φυσική». Δουλεύει και για πηγές dc ! Στη πραγματικότητα ένας μετασχηματιστής καίγεται στο dc .

- **Ιδανικός μετασχηματιστής με αυτεπαγωγές.**


Ο άλλος τρόπος μοντελοποίησης μετασχηματιστή χωρίς απώλειες είναι με τη χρήση δυο αυτεπαγωγών με συντελεστή σύζευξης ίσον με 1 . Ισχύουν οι σχέσεις,

$$L_{SEC} = L_{PRI} * N^2$$

Όπου  $L_{SEC}$ ,  $L_{PRI}$  οι αυτεπαγωγές πρωτεύοντος και δευτερεύοντος και  $N$  όπως και στη προηγούμενη περίπτωση.

Αν  $N=5$  και η αυτεπαγωγή του πρωτεύοντος είναι 500 mH τότε έχουμε το Σχήμα 18.

$$LSEC = (50 \times 10^{-3}) \times 5^2 \text{ H} = 1,25 \text{ H}.$$



Σχήμα 18

Οι εντολές είναι,

LPRI 30 40 50E-3

LSEC 50 60 1.25

KPRI-SEC LPRI LSEC 1.0

Σημειώστε ότι ο πρώτος κόμβος της κάθε αυτεπαγωγής είναι ο κόμβος ο οποίος είναι συνδεδεμένος με το άκρο που δηλώνεται στο σχεδιάγραμμα με την μαύρη κουκίδα. Αυτό εξασφαλίζει την κατάλληλη πολικότητα για την επαγόμενη τάση στο

δευτερεύον. Στην περίπτωση με αυτεπαγωγές ο μετασχηματιστής δεν «λειτουργεί» με τάση dc .


- **Μοντέλο διακόπτη.**

Η παραλλαγή SPICE2 δεν έχει διακόπτες ενώ η παραλλαγή SPICE3 έχει. Σε κάθε περίπτωση μπορεί κάποιος να φτιάξει μοντέλο διακόπτη αξιοποιώντας μια μη γραμμικές (πολυνυμική) πηγή ρεύματος ελεγχόμενη από τάση ή ρεύμα. Θα εξετάσουμε τη περίπτωση διακόπτη με πηγή ελεγχόμενη από τάση (VCCS). Το τέχνασμα συνίσταται στο να φτιάξει κάποιος μια VCCS της οποίας το ρεύμα είναι το γινόμενο δυο τάσεων,

Η μια τάση είναι μια βηματική συνάρτηση από 0 V σε 1 V, που παριστάνει κλείσιμο διακόπτη (διακόπτης ΕΝΤΟΣ, ON), και από 1 V σε 0 V, για διακόπτη που ανοίγει (διακόπτης ΕΚΤΟΣ, OFF). Αυτό μπορεί να επιτευχθεί εύκολα με μια ανεξάρτητη πηγή PULSE ή μια πηγή PWL .

Η άλλη τάση είναι η τάση στα άκρα της ίδιας της VCCS . Υποθέτοντας διαγωγιμότητα με πολύ μεγάλη τιμή, πχ  $1 \times 10^6$  siemens (S), τότε η αντίσταση του διακόπτη στη κατάσταση ΕΝΤΟΣ θα είναι μόνο  $1 \times 10^{-6} \Omega$  .


Η λειτουργία του διακόπτη είναι προφανής.



Σχήμα 19

Παράδειγμα:

Έχουμε το κύκλωμα του Σχήματος 19 και το κύκλωμα του Σχήματος 20 όπου έχει αντικατασταθεί ο διακόπτης με την VCCS GSWITCH μαζί με την ανεξάρτητη πηγή τάσης PULSE (VPULSE) η οποία «λειτουργεί» τον διακόπτη. Επειδή χρειάζεται πηγή τάσης VPULSE να είναι μέρος «αποδεκτού» κυκλώματος, συνδέεται στα άκρα της ο ανενεργός αντιστάτης RPULSE = 1 Ω .



Σχήμα 20

Το αρχείο .CIR για αυτό το κύκλωμα είναι,


```
SWITCH.CIR (USE OF VCCS TO MAKE A SWITCH)
VPULSE 100 0 PULSE(0 1 1 1M 1M 1)
RPULSE 100 0 1
VL 1 0 10
RL 2 3 20
RMID 3 0 5
RR 3 4 5
VR 4 0 2
GSWITCH 1 2 POLY(2) 100 0 1 2 0 0 0 0 1E6
.TRAN 0.1 2.5
.END
```

Ο παλμός (pulse) ξεκινά, με καθυστέρηση 1 s, από 0 V και φτάνει τη τιμή 1 V . Ο χρόνος ανόδου και καθόδου του παλμού είναι 1 ms και το εύρος (η διάρκειά) του

είναι 1 s . Ο διακόπτης VCCS GSWITCH είναι συνδεδεμένος μεταξύ των κόμβων 1 και 2 και ελέγχεται από την τάση V(100,0) (τον παλμό) και την τάση V(1,2) (στις παρενθέσεις φαίνονται οι κόμβοι όπου είναι οι συνδέσεις). Ο συντελεστής p4 του διακόπτη GSWITCH ορίζεται ίσος με  $1 \times 10^6$  S που αντιστοιχεί σε αντίσταση επαφής  $10^{-6} \Omega$  .

- Διακόπτης τριφασικής ισχύος.

Η διάταξη που περιγράφουμε θέτει ENTOΣ τριφασικά φορτία σε τριφασική παροχή ισχύος, σε οποιαδήποτε χρονική στιγμή του κύκλου (φάσης) των  $360^\circ$  . Οι τρεις «φάσεις» σημειώνονται με τα A, B, C . Θυμίζουμε ότι οι τρεις «φάσεις» (τάσεις) έχουν μεταξύ τους διαφορές φάσεις  $120^\circ$  . Η διάταξη φαίνεται στο Σχήμα 21 .



Σχήμα 21

Οι εντολές είναι,

```
3-PHASE.CIR  GENERATING SWITCHED 3-PHASE POWER
*  VCCS ARE USED TO SWITCH ALL THREE PHASES ON AT THE
*  SAMETIME
*  THE THREE FREE-RUNNING PHASES ARE CREATED BELOW
VA 1 0 SIN(0 311.1 50 0)
VB 2 0 SIN(0 311.1 50 6.66667M)
VC 3 0 SIN(0 311.1 50 13.3333 M)
RA 1 0 1
RB 2 0 1
RC 3 0 1
*  THE SWITCH CONTROL VOLTAGE FOR ALL 3 PHASES FOLLOWS
VSW 100 0 PULSE(0 1 16M 1N)
RSW 100 0 1
*  THREE VCCS SWITCHES FOLLOW
GA 1 11 POLY(2) 100 0 1 11 0 0 0 0 1E6
GB 2 12 POLY(2) 100 0 2 12 0 0 0 0 1E6
```

```
GC 3 13 POLY(2) 100 0 3 13 0 0 0 0 1E6
RGA 11 0 1
RGB 12 0 1
RGC 13 0 1
.TRAN 1M 60M 20M
.END
```

Η συχνότητα είναι 50 Hz άρα η περίοδος είναι 20 ms .Το πλάτος της τάσης είναι 311,1 V (ενεργός τιμή 220 V) .

Βλέπε σχετικά κεφάλαια για τη χρήση των διαφόρων εντολών.

## Διαφορές μεταξύ SPICE και PSpice.

Οι διαφορές μεταξύ SPICE (SPICE2G.6 του Πανεπιστημίου Berkeley) και του PSpice (της εταιρίας MicroSim) είναι δυο ειδών. Η πρώτη κατηγορία περιλαμβάνει τις διαφορές που εμφανίζονται όταν κάποιος χρησιμοποιεί το PSpice για να προσομοιώσει κάποιο κύκλωμα που περιγράφεται από ένα αρχείο εισόδου που φτιάχτηκε αρχικά για το SPICE. Η δεύτερη κατηγορία περιλαμβάνει τις διαφορές που συναντά κάποιος κατά την αντίθετη κατεύθυνση, δηλαδή όταν χρησιμοποιεί το SPICE για να τρέξει αρχείο εισόδου φτιαγμένο αρχικά για το PSpice. Θα αναφερθούμε σε μερικές μόνο διαφορές. Για περισσότερες λεπτομέρειες δείτε τα σχετικά φυλλάδια της MicroSim.

Με τις παρακάτω εξαιρέσεις κάθε αρχείο εισόδου που έχει φτιαχτεί για το SPICE μπορεί να χρησιμοποιηθεί για το PSpice.

1. Το PSpice δεν υποστηρίζει τα εξής πεδία στην εντολή .OPTION: LIMTIM, LVLCOD, METHOD, MAXORD, LVLTIM και ITL3.
2. Το PSpice δεν υποστηρίζει την δυνατότητα IN = για την εντολή .WIDTH .
3. Οι συντελεστές τάσης για πυκνωτές και οι συντελεστές ρεύματος για επαγωγές πρέπει να εισαχθούν στην εντολή .MODEL και όχι στην εντολή που ορίζει τη διάταξη.


Γενικώς το SPICE μπορεί να διαβάζει και να τρέχει αρχεία εισόδου που φτιάχτηκαν για το PSpice. Όμως μερικές δυνατότητες που έχει το PSpice δεν υποστηρίζονται από το SPICE. Κάθε αρχείο εισόδου που χρησιμοποιεί κάποια από τις κατωτέρω δυνατότητες πρέπει να τροποποιηθεί πριν χρησιμοποιηθεί από το SPICE.

1. Το SPICE δεν υποστηρίζει διακόπτες ελεγχόμενους από τάση ή ρεύμα.
2. Το SPICE δεν υποστηρίζει μοντέλα αντιστατών, αυτεπαγωγών, και πυκνωτών, όπως αυτά που είναι σε συνδυασμό με την εντολή .STEP.
3. Το SPICE δεν επιτρέπει λογαριθμικές σαρώσεις τάσεων και ρευμάτων dc .
4. Το SPICE δεν υποστηρίζει σαρώσεις παραμέτρων στην εντολή .MODEL .
5. Το SPICE δεν υποστηρίζει την εντολή .LIB .
6. Όλοι οι χαρακτήρες στο αρχείο εισόδου για το SPICE πρέπει να είναι με κεφαλαία.



## ΜΕΡΙΚΑ ΕΝΔΕΙΚΤΙΚΑ ΑΠΟΤΕΛΕΣΜΑΤΑ ΑΝΑΛΥΣΗΣ

### ΠΑΡΑΔΕΙΓΜΑ 1




Παράδειγμα 1

#### CHARACTERISTIC OF A DIODE


```
VD 10 0
D 10 20 PLAIN
V-ID 20 0 0
.DC VD 0.6 0.81 0.01
.MODEL PLAIN D
.PLOT DC I(V-ID) (0, 0.4)
.PROBE
.OPTIONS NOPAGE
.END
```

## Οδηγίες προσομοίωσης και παρουσίασης των αποτελεσμάτων:


Υποθέτουμε ότι η παραπάνω λίστα περιγραφής του κυκλώματος έχει γραφεί και αποθηκευτεί σε ένα αρχείο, έστω par1.cir. Για να γίνει προσομοίωση του κυκλώματος παίρνουμε αυτό το αρχείο και το σέρνουμε (drag and drop) πάνω στο εικονίδιο PSpice AD όπως φαίνεται στην παρακάτω εικόνα.




Σχεδόν αμέσως (για τα απλά κυκλώματα) παίρνουμε το μήνυμα ότι η προσομοίωση έχει ολοκληρωθεί.



Αν δούμε τον κατάλογο στον οποίο πριν βρισκόταν το par1.cir θα παρατηρήσουμε δύο νέα αρχεία, τα par1.dat και το par1.out το οποίο περιέχει τα αποτελέσματα που ζητήσαμε.



Αν ανοίξουμε το par1.out με έναν editor όπως το notepad ή το word θα δούμε όλες τις πληροφορίες που σχετίζονται με το κύκλωμα και τον τύπο της εξομοίωσης που ζητήσαμε.



Αναλυτικά, για το παράδειγμα 1 θα δούμε τα παρακάτω:

\*\*\*\* 09/02/102 15:36:38 \*\*\*\*\* NT Evaluation PSpice (July 1997) \*\*\*\*\*

# CHARACTERISTIC OF A DIODE

## \*\*\*\* CIRCUIT DESCRIPTION

\*\*\*\*\*

```
VD 10 0
D 10 20 PLAIN
V-ID 20 0 0
.DC VD 0.6 0.81 .01
.MODEL PLAIN D
.PLOT DC I(V-ID) (0, 0.4)
.PROBE
.OPTIONS NOPAGE
.END
```

## \*\*\*\* Diode MODEL PARAMETERS

```
PLAIN
IS 10.000000E-15
```

\*\*\*\* DC TRANSFER CURVES            TEMPERATURE = 27.000 DEG C


```
VD        I(V-ID)
(*)----- 0.0000E+00 1.0000E-01 2.0000E-01 3.0000E-01 4.0000E-01

-----
6.000E-01 1.188E-04 * . . . .
6.100E-01 1.749E-04 * . . . .
6.200E-01 2.574E-04 * . . . .
6.300E-01 3.789E-04 * . . . .
6.400E-01 5.578E-04 * . . . .
6.500E-01 8.211E-04 * . . . .
6.600E-01 1.209E-03 * . . . .
6.700E-01 1.779E-03 * . . . .
6.800E-01 2.619E-03 * . . . .
6.900E-01 3.855E-03 * . . . .
7.000E-01 5.675E-03 * . . . .
7.100E-01 8.353E-03 * . . . .
7.200E-01 1.230E-02 * . . . .
7.300E-01 1.810E-02 * . . . .
7.400E-01 2.664E-02 * . . . .
7.500E-01 3.922E-02 * . . . .
7.600E-01 5.773E-02 * . . . .
7.700E-01 8.498E-02 * . . . .
7.800E-01 1.251E-01 . * . . .
7.900E-01 1.841E-01 . . * . .
8.000E-01 2.711E-01 . . . * .
8.100E-01 3.990E-01 . . . . *
```


JOB CONCLUDED

TOTAL JOB TIME        .01


Ένας άλλος πίο όμορφος και σύγχρονος τρόπος για να δούμε τα αποτελέσματα είναι με την χρήση του Probe του PSpice. Μετά λοιπόν το τέλος της εξομοίωσης από το PSpice AD επιλέγουμε από το μενού File... Run Probe.




Στην αρχή εμφανίζεται μία κενή γραφική παράσταση. Για να προσθέσουμε τα στοιχεία που θέλουμε να δούμε πατάμε το κουμπί add trace.




Στη συνέχεια επιλέγουμε τα μεγέθη που θέλουμε από τη λίστα



Και έχουμε την ζητούμενη γραφική παράσταση.



## ΠΑΡΑΔΕΙΓΜΑ 2



Παράδειγμα 2

7404 TTL INVERTER, SWEPT DC INPUT

VCC 4 0 DC 5

VIN 1 0

RBQ1 4 2 4K

RCQ2 4 5 1.6K

REQ2 6 0 1K

RCQ3 4 7 130

RPULLUP 4 9 2K

DOUTPUT 8 9 DI-MOD

Q1 3 2 1 Q-MOD

Q2 5 3 6 Q-MOD

Q3 7 5 8 Q-MOD

Q4 9 6 0 Q-MOD

.MODEL DI-MOD D

.MODEL Q-MOD NPN(BF=50)


.OPTIONS NOPAGE

.DC VIN 1.3 1.7 0.01


.PROBE V(9) V(1)

.PLOT DC V(9,0) (0,5) V(1,0) (1,2)

.END



### ΠΑΡΑΔΕΙΓΜΑ 3



Παράδειγμα 3

#### DOUBLE TUNED OVER COUPLED RF TRANSFORMER

```

ISOURCE 0 5 AC 0.5M
RSOURCE 5 0 100K
CPRIM 5 0 316.63P
RPRIM 5 6 2
LPRIM 6 0 20U
KP-S LPRIM LSEC 0.10
LSEC 1 3 20U
RSEC 1 2 2
CSEC 2 3 316.63P
RL 2 3 30K
RDCPATH 3 0 1T

```



```
.AC LIN 41 1.8MEG 2.2MEG
*.PLOT AC VM(2,3) (0.3,5)
.OPTIONS NOPAGE
.PROBE
.END
```

```
**** 09/02/102 16:04:52 ***** NT Evaluation PSpice (July 1997) *****
```

```
DOUBLE TUNED OVER COUPLED RF TRANSFORMER
```

```
**** CIRCUIT DESCRIPTION
```

```
*****
```

```
ISOURCE 0 5 AC 0.5M
RSOURCE 5 0 100K
CPRIM 5 0 316.63P
RPRIM 5 6 2
LPRIM 6 0 20U
KP-S LPRIM LSEC 0.10
LSEC 1 3 20U
RSEC 1 2 2
CSEC 2 3 316.63P
RL 2 3 30K
RDCPATH 3 0 1T
.AC LIN 41 1.8MEG 2.2MEG
*.PLOT AC VM(2,3) (0.3,5)
.OPTIONS NOPAGE
.PROBE
.END
```

```
**** SMALL SIGNAL BIAS SOLUTION TEMPERATURE = 27.000 DEG C
```

```
NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE
```

```
( 1) 0.0000 ( 2) 0.0000 ( 3) 0.0000 ( 5) 0.0000
( 6) 0.0000
```


```
VOLTAGE SOURCE CURRENTS
NAME CURRENT
```

```
TOTAL POWER DISSIPATION 0.00E+00 WATTS
```

```
JOB CONCLUDED
```

```
TOTAL JOB TIME .05
```

## ΠΑΡΑΔΕΙΓΜΑ 4



Παράδειγμα 4

BJT DIFFERENTIAL AMPLIFIER, AC ANALYSIS

VSIG 3 0 AC 1

VCC 1 0 DC 12

VEE 5 0 -12

Q1 2 3 4 APPLE

Q2 6 0 4 APPLE

.MODEL APPLE NPN(BF=60 CJC=16P CJE=30P)

RC1 1 2 5K

RC2 1 6 5K

RE 4 5 4.8K

.AC DEC 5 100 1G


.PLOT AC VDB(6) (-10,50) VP(6) (-90,0)

.OPTIONS NOPAGE


.OP

.PROBE

.END



## ΠΑΡΑΔΕΙΓΜΑ 5



Παράδειγμα 5

# PIECEWISE LINEAR VOLTAGE INTO R-C LOWPASS FILTER

```
VIN 5 0 PWL(0 0 1M 1 2M 0 3M 1 4M 0)
```

```
R 5 8 1K
```

```
C 8 0 2U
```


```
.TRAN .1M 4M
```

```
.PLOT TRAN V(5) (0,1) V(8) (0,.5)
```


```
.OPTIONS NOPAGE
```

```
.PROBE
```

```
.END
```



## ΠΑΡΑΔΕΙΓΜΑ 6





Παράδειγμα 6


```

½ WAVE POWER SUPPLY
VAC 10 0 SIN(0 100 50)
* H VAC EINAI 100V 50Hz
VSENSE 10 12 0
D 12 15 MODA
.MODEL MODA D(N=0.001 RS=2)
CFILTER 15 0 40U
RLOAD 15 0 1K
.TRAN1M 40M
.PLOT TRAN V(15) (0,100) I(VSENSE) (0,1.5)
.OPTIONS NOPAGE
.PROBE
.END

```



## ΠΑΡΑΔΕΙΓΜΑ 7




Παράδειγμα 7


```

TTL 7404 INVERTER GATE, PULSE INPUT
VCC 4 0 5
VIN 1 0 PULSE(0 4 10N 2N 2N 20N 48N)
RBQ1 4 2 4K
RCQ2 4 5 1.6K
REQ2 6 0 1K
RCQ3 4 7 130
RPULLUP 4 9 2K
DOUTPUT 8 9 DMOD1
Q1 3 2 1 MOD1
Q2 5 3 6 MOD1
Q3 7 5 8 MOD1
Q4 9 6 0 MOD1
.WIDTH OUT = 80
.MODEL MOD1 NPN(BF=50 BR=0.1 RB=70 RC=40 TF=.1N TR=10N
+ CJE=0.9P CJC=1.5P CCS=1P VA=50)
.MODEL DMOD1 D
.TRAN 3N 105N
*.PLOT TRAN V(9) (0,12) V(1) (-5,5)
.PROBE
.OPTIONS NOPAGE
.END

```



## ΠΑΡΑΔΕΙΓΜΑ 8



Παράδειγμα 8

# ASTABLE BJT CIRCUIT

\* ΣΤΗΝ ΕΞΟΔΟ ΕΧΟΥΜΕ ΤΕΤΡΑΓΩΝΙΚΟ ΣΗΜΑ

Q1 1 2 0 MOD1

Q2 4 5 0 MOD1

RC1 7 1 1K

RB1 7 2 50K

RC2 7 4 1K

RB2 7 5 50K

CC1B2 1 5 20U

CC2B1 4 2 20U

VCC 7 0 6

.MODEL MOD1 NPN(TF=10N CJC=1P CJE=1P)

.TRAN 0.05 2.5 UIC


.IC V(1)=0.1 V(4)=6 V(5)=-6 V(2)=0.8 V(7)=6

.PLOT TRAN V(1) (0,12) V(2) (-13,1)

.OPTIONS NOPAGE RELTOL=0.01


.PROBE

.END





## ΠΑΡΑΔΕΙΓΜΑ 9



Παράδειγμα 9

AMPLITUDE MODULATION USING A POLYNOMIAL VCVS

V-CAR 10 0 SIN(0 10 10K)

\* Η V-CAR ΕΧΕΙ ΠΛΑΤΟΣ 10VP ΚΑΙ ΕΧΕΙ ΦΕΡΟΝ ΗΜΙΤΟΝΙΚΟ 10KHZ

R-CAR 10 0 1

V-MOD 20 0 SIN(1 1 2K 200U)

\* Η V-MOD ΕΧΕΙ ΠΛΑΤΟΣ 1VDC + 1VP ΗΜΙΤΟΝΙΚΟ ΣΤΑ 2KHZ, ΜΕ ΚΑΘΥΣΤΕΡΗΣΗ 200U, ΚΑΙ ΔΙΑΜΟΡΦΩΜΕΝΗ

R-MOD 20 0 1

E-AM 30 0 POLY(2) 10 0 20 0 0 0 0 0 1

R-AM 30 0 1


.TRAN 10U 1M

.OPTIONS NOPAGE


.PLOT TRAN V(20) V(30)

.PROBE

.END



## ΠΑΡΑΔΕΙΓΜΑ 10



Παράδειγμα 10

½ WAVE RECTIFIER WITH FOURIER ANALYSIS

VAC 2 0 SIN(0 1.0 100)

D 2 4 IDEAL

RL 4 0 1K

.MODEL IDEAL D(N=0.001)

.TRAN 0.25M 10M


.FOUR 100 V(4,0)

.PLOT TRAN V(4,0) V(2,0)


.OPTIONS NOPAGE

.PROBE

.END



## ΠΑΡΑΔΕΙΓΜΑ 11



Παράδειγμα 11

### DIFFERENTIAL AMPLIFIER WITH SENSITIVITY ANALYSIS

```
VCC 1 0 14
VEE 2 0 -9
VIN1 3 0 AC 1M
Q1 4 3 5 MOD1
RC1 1 4 10K
RE 5 2 4.3K
VIN2 7 0 0
Q2 6 7 5 MOD1
RC2 1 6 10K
.OP
.SENS V(4)
.MODEL MOD1 NPN(BF=75)
.OPTIONS NOPAGE
.END
```

\*\*\*\* 09/02/102 16:25:18 \*\*\*\*\* NT Evaluation PSpice (July 1997) \*\*\*\*\*

#### DIFFERENTIAL AMPLIFIER WITH SENSITIVITY ANALYSIS

#### \*\*\*\* CIRCUIT DESCRIPTION

\*\*\*\*\*

```
VCC 1 0 14
VEE 2 0 -9
VIN1 3 0 AC 1M
Q1 4 3 5 MOD1
RC1 1 4 10K
RE 5 2 4.3K
VIN2 7 0 0
```

```

Q2 6 7 5 MOD1
RC2 1 6 10K
.OP
.SENS V(4)
.MODEL MOD1 NPN(BF=75)
.OPTIONS NOPAGE
.END

```

\*\*\*\* BJT MODEL PARAMETERS

```

MOD1
NPN
IS 100.000000E-18
BF 75
NF 1
BR 1
NR 1

```

\*\*\*\* SMALL SIGNAL BIAS SOLUTION TEMPERATURE = 27.000 DEG C

NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE NODE VOLTAGE

```

( 1) 14.0000 ( 2) -9.0000 ( 3) 0.0000 ( 4) 4.5593
( 5) -.7727 ( 6) 4.5593 ( 7) 0.0000

```

VOLTAGE SOURCE CURRENTS

NAME CURRENT

```

VCC -1.888E-03
VEE 1.913E-03
VIN1 -1.259E-05
VIN2 -1.259E-05

```

TOTAL POWER DISSIPATION 4.37E-02 WATTS

\*\*\*\* OPERATING POINT INFORMATION TEMPERATURE = 27.000 DEG C

\*\*\*\* BIPOLAR JUNCTION TRANSISTORS

NAME	Q1	Q2
MODEL	MOD1	MOD1
IB	1.26E-05	1.26E-05
IC	9.44E-04	9.44E-04
VBE	7.73E-01	7.73E-01
VBC	-4.56E+00	-4.56E+00
VCE	5.33E+00	5.33E+00
BETADC	7.50E+01	7.50E+01
GM	3.65E-02	3.65E-02
RPI	2.05E+03	2.05E+03
RX	0.00E+00	0.00E+00
RO	1.00E+12	1.00E+12
CBE	0.00E+00	0.00E+00
CBC	0.00E+00	0.00E+00
CJS	0.00E+00	0.00E+00
BETAAC	7.50E+01	7.50E+01
CBX	0.00E+00	0.00E+00
FT	5.81E+17	5.81E+17

\*\*\*\* DC SENSITIVITY ANALYSIS      TEMPERATURE = 27.000 DEG C

DC SENSITIVITIES OF OUTPUT V(4)

ELEMENT NAME	ELEMENT VALUE (VOLTS/UNIT)	ELEMENT SENSITIVITY (VOLTS/PERCENT)	NORMALIZED SENSITIVITY
RC1	1.000E+04	-9.441E-04	-9.441E-02
RE	4.300E+03	2.189E-03	9.411E-02
RC2	1.000E+04	-4.644E-12	-4.644E-10
VCC	1.400E+01	1.000E+00	1.400E-01
VEE	-9.000E+00	1.144E+00	-1.030E-01
VIN1	0.000E+00	-1.831E+02	-0.000E+00
VIN2	0.000E+00	1.819E+02	0.000E+00
Q1			
RB	0.000E+00	0.000E+00	0.000E+00
RC	0.000E+00	0.000E+00	0.000E+00
RE	0.000E+00	0.000E+00	0.000E+00
BF	7.500E+01	-8.255E-04	-6.192E-04
ISE	0.000E+00	0.000E+00	0.000E+00
BR	1.000E+00	1.000E-12	1.000E-14
ISC	0.000E+00	0.000E+00	0.000E+00
IS	1.000E-16	-4.735E+16	-4.735E-02
NE	1.500E+00	0.000E+00	0.000E+00
NC	2.000E+00	0.000E+00	0.000E+00
IKF	0.000E+00	0.000E+00	0.000E+00
IKR	0.000E+00	0.000E+00	0.000E+00
VAF	0.000E+00	0.000E+00	0.000E+00
VAR	0.000E+00	0.000E+00	0.000E+00
Q2			
RB	0.000E+00	0.000E+00	0.000E+00
RC	0.000E+00	0.000E+00	0.000E+00
RE	0.000E+00	0.000E+00	0.000E+00
BF	7.500E+01	-8.255E-04	-6.192E-04
ISE	0.000E+00	0.000E+00	0.000E+00
BR	1.000E+00	4.919E-21	4.919E-23
ISC	0.000E+00	0.000E+00	0.000E+00
IS	1.000E-16	4.706E+16	4.706E-02
NE	1.500E+00	0.000E+00	0.000E+00
NC	2.000E+00	0.000E+00	0.000E+00
IKF	0.000E+00	0.000E+00	0.000E+00
IKR	0.000E+00	0.000E+00	0.000E+00
VAF	0.000E+00	0.000E+00	0.000E+00
VAR	0.000E+00	0.000E+00	0.000E+00

JOB CONCLUDED

TOTAL JOB TIME      .02

# SPICE ΜΕ ΠΡΟΧΩΡΗΜΕΝΑ ΓΡΑΦΙΚΑ GUI (Graphical User Interfaces)

## Γενικά

Εκτός από τον κλασικό τρόπο χρήσης του PSpice (SPICE) θα αναφερθούμε σε έναν από τους πιο σύγχρονους τρόπους χρήσης που έχει, στην περίπτωσή μας, αναπτύξει η MicroSim. Για το σκοπό αυτό γίνεται χρήση των GUI (Graphical User Interfaces) τα οποία επιτρέπουν τη σχεδίαση του κυκλώματος στην οθόνη του υπολογιστή. Η σχεδίαση γίνεται με το πρόγραμμα Schematics μέσα από ένα περιβάλλον εργασίας όπου τα επιμέρους στοιχεία του κυκλώματος επιλέγονται ως σύμβολα και συνδέονται μεταξύ τους όπως θα γινόταν στην περίπτωση ενός πραγματικού κυκλώματος. Μάλιστα, με χρήση του mouse η διαδικασία είναι αρκετά απλή και διευκολύνει το χρήστη σε μεγάλο βαθμό. Στη συνέχεια, το κύκλωμα μπορεί να αποθηκευτεί ως αρχείο τύπου .SCH με επιλεγόμενο όνομα. Ένα προϋπάρχον κύκλωμα μπορεί να φορτωθεί για περαιτέρω επεξεργασία χρησιμοποιώντας την εντολή file/open . Ο χρήστης μπορεί να ορίσει το είδος της ανάλυσης με την εντολή analysis/setup ενώ με την εντολή analysis/simulate μπορεί να πραγματοποιήσει την ανάλυση αυτή. Αφού τελειώσει η ανάλυση, το πρόγραμμα PROBE κατασκευάζει τις γραφικές παραστάσεις, όπως γίνεται με τον κλασικό τρόπο. Τα διαγράμματα αφορούν στα ηλεκτρικά μεγέθη που αντιστοιχούν στο κύκλωμα και χαρακτηρίζονται από δείκτες που τίθενται αυτόματα, αν και μπορούν αλλάξουν από τον χρήστη. Ο αναφερόμενος τρόπος προσομοίωσης είναι αρκετά εποπτικός και προσφέρεται για εύκολη πραγματοποίηση μετατροπών-αλλαγών του κυκλώματος, εκτύπωση και ενσωμάτωσή του σε εργασίες.

## Διαδοχικά βήματα προσομοίωσης

Η προσομοίωση με χρήση GUI πραγματοποιείται ακολουθώντας συγκεκριμένα διαδοχικά βήματα στο περιβάλλον του προγράμματος Schematics. Εκτελώντας το πρόγραμμα αυτό «ανοίγει» ένας χώρος εργασίας κατάλληλος για τη σχεδίαση με χρήση συμβόλων. Παρακάτω παρουσιάζονται αναλυτικά τα βήματα της διαδικασίας που πρέπει κανείς να ακολουθήσει:

### 1. Σχεδίαση του ηλεκτρονικού κυκλώματος

#### α) Τοποθέτηση στοιχείων

Η επιλογή και η σύνδεση των διαφόρων στοιχείων (πηγών και εξαρτημάτων) του κυκλώματος γίνεται με έναν τρόπο ανάλογο με αυτόν που θα εφαρμόζαμε σε πραγματικό εργαστήριο. Πιο συγκεκριμένα, με το mouse και την εντολή draw/Get New Part (ή CTRL+G) για νέο στοιχείο ή draw/Place Part (ή CTRL+P) για ήδη χρησιμοποιημένο, επιλέγουμε από τη βιβλιοθήκη (part browser basic) διάφορα στοιχεία και τα τοποθετούμε στο χώρο εργασίας. Ο προσανατολισμός του κάθε εξαρτήματος μπορεί να αλλάξει κατά 90° με CTRL+R.

#### β) Ηλεκτρικές συνδέσεις

Η σύνδεση (διασυνδέωση) μεταξύ των εξαρτημάτων γίνεται με την εντολή Draw Wire (εικονίδιο «μολύβι») και χρήση του mouse. Οι κόμβοι του

κυκλώματος δημιουργούνται αυτόματα. Είναι απαραίτητο επίσης να τοποθετηθεί το σύμβολο το κόμβου αναφοράς AGND (analogue ground). Αν μετά την ολοκλήρωση των συνδέσεων γίνει κάποια μετακίνηση στοιχείου, η υπάρχουσα σύνδεση του διατηρείται, όπως και ο τυχόν γειτονικός κόμβος. Για το λόγο αυτό, στο βήμα 1.1 δεν είναι απαραίτητο τα στοιχεία να τοποθετηθούν στις επιθυμητές θέσεις από την αρχή αλλά μπορούν να διευθετηθούν σε μεταγενέστερο χρόνο. Σημειώνεται ότι στο χώρο εργασίας μπορεί κανείς να σχεδιάσει πολλά κυκλώματα με μικρές παραλλαγές ενός βασικού με σκοπό να μελετήσει τις προκύπτουσες διαφορές. Στην περίπτωση αυτή η προσομοίωση εκτελείται ταυτόχρονα σε όλα αυτά τα επιμέρους κυκλώματα, αρκεί το κάθε ένα από αυτά να είναι πλήρες από πλευράς συνδέσεων και χωρίς σφάλματα.

## 2. Ορισμός χαρακτηριστικών των στοιχείων

### α) Συμβολικά ονόματα

Για κάθε στοιχείο που τοποθετείται τίθεται από το πρόγραμμα ένα προεπιλεγμένο όνομα με δείκτη αύξουσας αρίθμησης. Όμως, μπορεί κανείς να αλλάξει την ονομασία, με διπλό κτύπημα του mouse στο σημείο αυτό, αρκεί να μην υπάρχει διπλή ονομασία.

### β) Τιμές στοιχείων

Το κάθε στοιχείο παίρνει επίσης αυτόματα μια προεπιλεγμένη τιμή η οποία στη συνέχεια μπορεί να αλλάξει με διπλό κτύπημα είτε πάνω στο σύμβολο είτε στο πλαίσιο με την προεπιλεγμένη τιμή.

### γ) Παράμετροι λειτουργίας πηγών

Ειδικά στις πηγές (τάσης ή ρεύματος) είναι αναγκαίο να ορισθούν πρόσθετες παράμετροι λειτουργίας, όπως το πλάτος, η συχνότητα, η χρονική καθυστέρηση, κ.α. ανάλογα με το είδος της πηγής. Αυτό μπορεί να γίνει με διπλό κτύπημα στο σύμβολο της πηγής, οπότε εμφανίζεται ένα παράθυρο που περιέχει όλες τις παραμέτρους. Η κάθε αλλαγή πρέπει να σώζεται προτού προχωρήσει κανείς στην επόμενη.

## 3. Προεπιλογές προσομοίωσης

### α) Είδος της ανάλυσης

Το είδος της ανάλυσης σχετίζεται με το αν κύκλωμα θα αναλυθεί στο πεδίο του χρόνου που αναφέρεται (time domain) ή στο πεδίο συχνοτήτων (frequency domain). Η επιλογή μπορεί να γίνει με την εντολή Analysis/Setup, όπου, στο παράθυρο που ανοίγει, πρέπει κανείς να σημειώσει (checking) στα αντίστοιχα κουτάκια δίπλα στα εικονίδια (π.χ. AC Sweep για ανάλυση στο πεδίο των συχνοτήτων, Transient για ανάλυση στο πεδίο του χρόνου) τις επιλογές του. Το εικονίδιο DC Sweep αναφέρεται σε μια σειρά από παραμέτρους που μπορεί κανείς επίσης να μεταβάλλει. Το Bias Point Detail είναι πάντα σημειωμένο διότι το σημείο ηρεμίας στα τρανζίστορ υπολογίζεται πάντοτε.

### β) Μεταβλητές και ακρίβεια ανάλυσης

Με double click στο κάθε επιλεγμένο εικονίδιο μπορεί κανείς να κάνει επιλογή της μεταβλητής παραμέτρου (π.χ. Voltage Source, Temperature), του αριθμού των τιμών σημείων υπολογισμού (Total Pts), του εύρους μεταβολής (Start value, End value) καθώς και το είδος της κλίμακας (Linear, Octal, Decade). Ειδικά στην επιλογή Transient πρέπει κανείς να δώσει τιμή Final time και Step Ceiling



οποσδήποτε, ενώ η τιμή Print Step δεν είναι απαραίτητη για απεικόνιση στην οθόνη.

γ) Ορισμός σημείων ελέγχου

Στο ηλεκτρονικό σχεδιάγραμμα μπορεί κανείς να τοποθετήσει τα Voltage/Level Markers ή Current Marker στα σημεία εκείνα όπου, μετά την εκτέλεση της προσομοίωσης, θα δοθούν οι γραφικές παραστάσεις της τάσης ή ρεύματος συναρτήσει του χρόνου ή της συχνότητας, ανάλογα με την επιλογή που έχει γίνει.

#### 4. Πραγματοποίηση προσομοίωσης

α) Εκτέλεση

Αποθηκεύουμε το κύκλωμα σε αρχείο και στη συνέχεια εκτελούμε την προσομοίωση με την εντολή Analysis/Simulate (ή F11).

β) Παρουσίαση αποτελεσμάτων

Τα αποτελέσματα της προσομοίωσης παρέχονται με τους παρακάτω τρόπους: Οι συνεχείς τάσεις και ρεύματα απεικονίζονται πάνω στο ηλεκτρονικό σχεδιάγραμμα με την εντολή Enable Bias Voltage Display ή Enable Bias Current Display με single click στα αντίστοιχα εικονίδια. Οι μεταβολές των μεγεθών (σύμφωνα με τα σημεία ελέγχου) απεικονίζονται μέσω του προγράμματος .Probe. Στο περιβάλλον αυτό μπορεί κανείς να κάνει διάφορες επιλογές στα διαγράμματα, όπως, απεικονιζόμενο μέγεθος, αλλαγές στις περιοχές απεικόνισης, ανάλυση FFT κ.α. Τέλος, σχετικά αριθμητικά αποτελέσματα δίνονται στο output file.

## Χρήση υποκυκλωμάτων

Στην περίπτωση που ένα (υπο)κύκλωμα εμφανίζεται περισσότερο από μια φορά (π.χ. κυκλώματα τελεστικού ενισχυτή ως βαθμίδες ενίσχυσης) σε ένα μεγαλύτερο κύκλωμα, με χρήση του PSpice μπορεί να εισαχθεί ως έτοιμη διάταξη όπου χρειάζεται. Παρακάτω αναλύεται η διαδικασία χρήσης υποκυκλωμάτων κατά τη σχεδίαση:

#### 1. Ορισμός πλαισίων υποκυκλωμάτων

Αρχικά είναι απαραίτητο, στο χώρο εργασίας όπου σχεδιάζουμε ή πρόκειται να σχεδιάσουμε το κυρίως κύκλωμα, να ορίσουμε και να τοποθετήσουμε τα πλαίσια (blocks) των υποκυκλωμάτων που θα εισάγουμε, αρκεί να έχουμε προηγουμένως σώσει με κάποιο όνομα το τρέχον κύκλωμα. Αυτό γίνεται με την εντολή draw/block μέσω της οποίας ορίζουμε όλα τα απαιτούμενα πλαίσια (που αρχικά είναι κενά) και τα τοποθετούμε στις επιθυμητές θέσεις. Το πρόγραμμα, αυτόματα, δίνει ονόματα στα πλαίσια (π.χ. HB1) τα οποία όμως μπορούν και να τροποποιηθούν από το χρήστη με το γνωστό τρόπο (διπλό κτύπημα στο όνομα). Επίσης, με διπλό κτύπημα στο κάθε πλαίσιο προκύπτει το παράθυρο Set Up Block στο οποίο δίνουμε το όνομα του υποκυκλώματος στη θέση filename. Μετά το OK το υποκύκλωμα πλέον έχει αποκτήσει το πλήρες όνομα που έχει την εξής δομή: «blockname.filename.sch». Για παράδειγμα, αν το πλαίσιο ονομάστηκε HB1 και το filename (του υποκυκλώματος) ήταν amplifier, τότε το πλήρες όνομα που θα αποκτήσει αυτόματα θα είναι: «HB1.amplifier.sch». Η πρόσβαση (ανάγνωση) του αρχείου αυτού γίνεται μόνο μέσα από το περιβάλλον

εργασίας του κυρίως κυκλώματος, όπως αναφέρεται παρακάτω (βήμα 4). Η σύνδεση του πλαισίου με το κυρίως κύκλωμα είναι απαραίτητη στη φάση αυτή και γίνεται με το γνωστό τρόπο, δηλαδή με διασυνδέωση, αλλά με τη διαφορά ότι σε κάθε σύνδεση εισόδου ή εξόδου το πρόγραμμα δίνει στα σημεία σύνδεσης ονόματα αυτόματα και με αυξανόμενη αρίθμηση (π.χ. P1, P2,...). Τα ονόματα αυτά είναι απαραίτητα και θα ληφθούν υπόψη στο βήμα 3. Φυσικά, με διπλό κτύπημα, μπορεί κανείς να τα αλλάξει δίνοντας διαφορετικά (μνημονικά) ονόματα.

## 2. Εισαγωγή του κάθε υποκυκλώματος

Το υποκύκλωμα είτε εισάγεται ως ένα προϋπάρχον είτε δημιουργείται εξ αρχής. Στην πρώτη περίπτωση, ανοίγει κανείς το αντίστοιχο file και αντιγράφει με μαρκάρισμα και αντιγραφή (copy) ένα μέρος ή όλο το ηλεκτρονικό σχέδιο που θα αποτελέσει υποκύκλωμα. Στη συνέχεια, στο χώρο εργασίας του κυρίως κυκλώματος, με διπλό κτύπημα στο αντίστοιχο πλαίσιο, μεταβαίνουμε στο χώρο εργασίας του υποκυκλώματος όπου γίνεται και η επικόλληση (paste). Το υποκύκλωμα πλέον έχει εισαχθεί και μπορούμε να βγούμε από το χώρο αυτό.

## 3. Συνδέσεις με το κυρίως κύκλωμα

Μετά τη δημιουργία των υποκυκλωμάτων είναι απαραίτητο να γίνουν οι συνδέσεις (δύο ή περισσότερες κατά κανόνα) με το κυρίως κύκλωμα. Αυτό γίνεται με διπλό κτύπημα στο κάθε πλαίσιο υποκυκλώματος, με αποτέλεσμα να μεταβούμε στον αντίστοιχο χώρο εργασίας. Εκεί, μπορεί κανείς να συνδέσει τα προβλεπόμενα σημεία (κόμβους ή ακροδέκτες) με τα αντίστοιχα σημεία του πλαισίου χρησιμοποιώντας τα ειδικά σύμβολα για το σκοπό αυτό που βρίσκονται στη βιβλιοθήκη συμβόλων και αναφέρονται ως IF\_IN και IF\_OUT για είσοδο και έξοδο αντίστοιχα. Τα ονόματα που θα δοθούν σε αυτά θα είναι απαραίτητα ταυτόσημα με τα αντίστοιχα του σημείου του πλαισίου.

## 4. Τροποποιήσεις

Στην περίπτωση που θέλουμε να τροποποιήσουμε κάποιο υποκύκλωμα πριν ή μετά την προσομοίωση, αυτό γίνεται με τον ακόλουθο τρόπο: Στο χώρο εργασίας του κυρίως κυκλώματος μεταβαίνουμε με το mouse στο πλαίσιο υποκυκλώματος που θέλουμε και με διπλό κτύπημα μεταβαίνουμε στο χώρο εργασίας του υποκυκλώματος. Εκεί, μπορούμε να πραγματοποιήσουμε οποιαδήποτε τροποποίηση και να το σώσουμε με το ίδιο όνομα. Η επαναφορά στο κυρίως κύκλωμα γίνεται μέσω του window (ή παραθύρου) στην μπάρα του χώρου εργασίας του Schematics.

# Εκτυπώσεις και δημιουργία αρχείων-εικόνων

Μετά από μια προσομοίωση μπορεί κανείς να εκτυπώσει ή να δημιουργήσει ηλεκτρονικά αρχεία εικόνων είτε με το σχεδιάγραμμα του κυκλώματος είτε με τα διαγράμματα αποτελεσμάτων. Παρακάτω περιγράφονται οι πιο γνωστοί τρόποι αυτών των διαδικασιών:

- α) Ένας απλός και αποτελεσματικός τρόπος είναι η χρήση της εντολής Print Screen (πλήκτρο Prt sc). Πιο συγκεκριμένα, με αυτήν την εντολή σώζουμε προσωρινά

στον buffer την εμφανιζόμενη στην οθόνη εικόνα. Στη συνέχεια στον κειμενογράφο που δουλεύουμε (π.χ. Word), εκτελούμε την εντολή Paste και ανακτούμε την εικόνα. Στη συνέχεια, με το δεξί πλήκτρο του mouse επιλέγουμε «εμφάνιση και αποκόπτουμε (crop) τις ανεπιθύμητες περιοχές ξεκινώντας από τις τέσσερις γωνίες. Στην περίπτωση που η αποκοπή αυτή δεν εκτελείται, πραγματοποιούμε περιστροφή της εικόνας όσες φορές χρειάζεται για να επανέλθει στην αρχική της θέση ( $4 \times 90^\circ$  ή  $2 \times 180^\circ$ ) και στη συνέχεια αποκόπτουμε με τον τρόπο που αναφέρθηκε (η διαδικασία περιστροφής επιτρέπει τη περαιτέρω επεξεργασία). Από το παράθυρο διαλόγου «γραμμής εργαλείων εικόνας» μπορούμε να επιλέξουμε κλίμακα του γκρι, συμπίεση κ.α.

- β) Με χρήση του προγράμματος Snagit επιλέγοντας μέρος της συνολικής εικόνας της οθόνης. Μέσω της εντολής CTRL+<SHIFT>+P, επιλέγουμε τα όρια της επιθυμητής εικόνας και με την εντολή Crop αποκόπτουμε και στη συνέχεια σώζουμε σε αρχείο του τύπου επιθυμούμε (π.χ. bmp).
- γ) Επειδή στο χώρο εργασίας του Schematics μπορεί να υπάρχουν περισσότερα του ενός κυκλώματα ή να μας ενδιαφέρει μόνο ένα τμήμα κυκλώματος, είναι επιθυμητή η επιλογή της αντίστοιχης περιοχής για δημιουργία εικόνας. Αυτό γίνεται με το αριστερό πλήκτρο του mouse (μαρκάρισμα της περιοχής). Στη συνέχεια, εκτελούμε την εντολή File/Export και δημιουργούμε αρχείο τύπου .DXF με το σχεδιάγραμμα που επιλέξαμε. Τέτοια αρχεία μπορούν να διαβαστούν από προγράμματα (όπως το Microsoft Visio) με σκοπό τη μετατροπή τους σε αρχεία τύπου .DWG για Autocad ή εικόνες τύπου .JIF, .JPG, .TIF, .EPS που με τη σειρά τους μπορούν να εισαχθούν σε κειμενογράφους (Word, Latex) ή ακόμη και στο Power Point για παρουσίαση. Ο τρόπος αυτός παρουσιάζει δυσκολίες στην περίπτωση του διαγράμματος αποτελεσμάτων (από το Probe) λόγω του μαύρου φόντου που διατηρείται. Αν πρόκειται για εκτύπωση, με την εντολή File/Print Preview βλέπει κανείς το διάγραμμα όπως αυτό πρόκειται να εκτυπωθεί και φυσικά στον εκτυπωτή το φόντο είναι λευκό.
- δ) Ένας ακόμη τρόπος είναι να εκτυπώσουμε το διάγραμμα, μέσω της εντολής File/Print, σε virtual printer που είναι ο Distiller και στη συνέχεια να σώσουμε το αρχείο σε μορφή .PDF και να το μετατρέψουμε στη συνέχεια σε έναν από τους παραπάνω τύπους. Επίσης ακολουθώντας τη μέθοδο (β) επιλέγουμε και σώζουμε το εμφανιζόμενο σχήμα ή διάγραμμα.
- ε) Υπάρχει επίσης δυνατότητα να μεταφέρει κανείς τα δεδομένα από τη γραφική παράσταση (ως x,y δεδομένα), όπως εμφανίζεται στο Probe, σε περιβάλλον επεξεργασίας (π.χ. Excel). Αυτό γίνεται με click στη μεταβλητή (χρωματίζεται), CTRL+C (copy) και στη συνέχεια CTRL+V (paste) στο αναφερόμενο περιβάλλον. Το διάγραμμα από το Probe μπορεί να μεταφερθεί ως εικόνα στο Word, πολύ εύκολα, ως εξής: στην πάνω μπάρα επιλέγουμε window και στη συνέχεια copy to clipboard. Στο Word με επιλογή paste το διάγραμμα μεταφέρεται αυτούσιο.


Παρακάτω, παραθέτουμε μια σειρά από παραδείγματα, μέσω των οποίων μπορεί κανείς να αποκτήσει εμπειρία σε βασικές διαδικασίες, ώστε στη συνέχεια να μάθει περισσότερες λεπτομέρειες μέσω των βοηθητικών αρχείων της εταιρίας. Είναι πολύ χρήσιμο να δοκιμάσει κανείς τα παραδείγματα στο δικό του υπολογιστή με διάφορες επιλογές παραμέτρων για βαθύτερη κατανόηση.

# 1. Σειρά αναλυτικών παραδειγμάτων

## Παράδειγμα 1.1 (παράλληλος συντονισμός)



Εκτέλεση του προγράμματος  
Schematics




Χώρος  
εργασίας




Εντολή: Get New Part (ή CTRL+G)


2




Επιλογή στοιχείου (στην προκειμένη περίπτωση εξάρτημα): αντιστάτης R



Μεταφορά και τοποθέτηση του στο χώρο εργασίας



Περιστροφή 90° με CTRL+R




Τοποθέτηση όλων των  
στοιχείων (της πηγής και των  
παθητικών εξαρτημάτων) με  
ορισμό των προβλεπόμενων  
τιμών







Εντολή Draw Wire

2





Διασυνδέωση





Δίνεται η εντολή: Analysis Setup


Σημειώθηκε η επιλογή AC Sweep (η Bias Point Detail είναι πάντοτε σημειωμένη)



Επιλέγονται:  
Σημεία υπολογισμού=100  
Αρχική συχνότητα=5 kHz  
Τελική συχνότητα=15 kHz



Γραμμική κλίμακα συχνότητας  
(AC Sweep type: linear)







Σώζεται το ηλεκτρονικό σχέδιο  
σε file του τύπου .sch


Στη συνέχεια πραγματοποιείται  
Η προσομοίωση






Διαγράμματα απόκρισης  
Συχνότητας στο σημείο  
Ελέγχου που επιλέχθηκε



Η κλίμακα συχνοτήτων  
εδώ είναι γραμμική, αλλά  
κατά κανόνα επιλέγουμε  
λογαριθμική  
(επιλογή: decade αντί linear)




Μέσα από το περιβάλλον  
του Probe επιλέγεται η  
απεικόνιση της τάσης ενός  
άλλου σημείου ελέγχου



## Παράδειγμα 1.2 (τελεστικός ενισχυτής)



Σημειώθηκε η επιλογή Transient  
για ανάλυση στο πεδίο χρόνου




Επιλέγονται:

Βήμα για εκτύπωση=4  $\mu$ s  
Τελικός χρόνος=4 ms



Βήμα μεταβολής=4  $\mu$ s

Παραλείπεται η αρχική  
μεταβατική απόκριση



---

### Παράδειγμα 1.3 (ενισχυτής τρανζίστορ κοινού εκπομπού)



**Transient**

Transient Analysis

Print Step: 2us

Final Time: 2ms

No-Print Delay:

Step Ceiling: 2us

☐ Detailed Bias Pt.

☐ Skip initial transient solution

Fourier Analysis


☐ Enable Fourier

Center Frequency:

Number of harmonics:



Output Vars.:

OK Cancel



## Παράδειγμα 1.4

(χαρακτηριστικές καμπύλες τρανζίστορ)



Επιλέγεται DC Sweep  
Με σκοπό τη μεταβολή  
της  $V_{CE}$



**DC Sweep**

Swept Var. Type

- ☒ Voltage Source
- ☐ Temperature
- ☐ Current Source
- ☐ Model Parameter
- ☐ Global Parameter

Sweep Type

- ☒ Linear
- ☐ Octave
- ☐ Decade
- ☐ Value List

Name: VCE

Model Type:

Model Name:

Param. Name:

Start Value: 0

End Value: 10

Increment: 0.01

Values:

Nested Sweep...

OK Cancel

Επιλέγονται για την πηγή VCE:

Αρχική τιμή=0  
Τελική τιμή=10 V  
Βήμα μεταβολής=0.01 V

Γραμμική κλίμακα (Linear)

**DC Nested Sweep**

Swept Var. Type

- ☐ Voltage Source
- ☐ Temperature
- ☒ Current Source
- ☐ Model Parameter
- ☐ Global Parameter

Sweep Type

- ☒ Linear
- ☐ Octave
- ☐ Decade
- ☐ Value List

Name: IB

Model Type:

Model Name:

Param. Name:

Start Value: 0

End Value: 50uA

Increment: 10uA

Values:

Main Sweep...

☒ Enable Nested Sweep


OK Cancel

Επιλέγονται για την πηγή IB  
(σε ένθετη μεταβολή, nested):

Αρχική τιμή=0  
Τελική τιμή=50  $\mu$ A  
Βήμα μεταβολής=10  $\mu$ A

Γραμμική κλίμακα (Linear)



Ενεργοποιείται η ένθετη  
Μεταβολή (Enable Nested  
Sweep)



Στο διάγραμμα του ρεύματος συλλέκτη,  $I_C$ , παρουσιάζεται το σμήνος καμπυλών για διάφορες τιμές του ρεύματος βάσης, όπως έχουν ορισθεί παραπάνω στο DC Nested Sweep

## Παράδειγμα 1.5

(χαρακτηριστική καμπύλη  $V_{CE}$ - $I_C$  με σταθερό  $I_B$ )



**DC Sweep** [X]

Swept Var. Type

- ☒ Voltage Source
- ☐ Temperature
- ☐ Current Source
- ☐ Model Parameter
- ☐ Global Parameter

Name:

Model Type:

Model Name:

Param. Name:

Sweep Type

- ☒ Linear
- ☐ Octave
- ☐ Decade
- ☐ Value List

Start Value:

End Value:

Increment:

Values:

Nested Sweep... OK Cancel

**DC Nested Sweep** [X]

Swept Var. Type

- ☐ Voltage Source
- ☒ Temperature
- ☐ Current Source
- ☐ Model Parameter
- ☐ Global Parameter

Name:

Model Type:

Model Name:

Param. Name:

Sweep Type

- ☐ Linear
- ☒ Octave
- ☐ Decade
- ☐ Value List

Start Value:


End Value:

Pts/Octave:

Values:

Main Sweep... ☐ Enable Nested Sweep

OK Cancel





## 2. Σειρά γενικότερων παραδειγμάτων

### Παράδειγμα 2.1



(κυκλώματα διαφορίσης και ολοκλήρωσης)

Τα κυκλώματα αυτά χρησιμοποιούν παθητικά στοιχεία (αντιστάτη και πυκνωτή) και πραγματοποιούν κατά προσέγγιση διαφορίση και ολοκλήρωση αντίστοιχα. Στο χώρο εργασίας έχουν σχεδιαστεί ανεξάρτητα το ένα από το άλλο και η προσομοίωση πραγματοποιείται ταυτόχρονα. Στις εισόδους εφαρμόζουμε περιοδικούς παλμούς πλάτους 10 V, περιόδου 100 ms και διάρκειας 20 ms μέσω των πηγών Vin-1 και Vin-2 του τύπου VPULSE. Στο analysis setup επιλέγουμε transient με βήμα 1 ms και τελικό χρόνο 150 ms.



## Παράδειγμα 2.2 (Βαθυπερατό και Υψιπερατό φίλτρο)



Τα παρακάτω κυκλώματα παρουσιάζουν απόκριση που εξαρτάται από τη συχνότητα του σήματος εισόδου. Η συμπεριφορά του κυκλώματος (αριστερά) επιτρέπει τη διέλευση κυρίως των υψηλών συχνοτήτων ενώ απορρίπτει τις χαμηλές, δηλαδή είναι υψιπερατό. Αντίθετα, το κύκλωμα (δεξιά) επιτρέπει τη διέλευση των χαμηλών κυρίως συχνοτήτων ενώ απορρίπτει τις υψηλές, δηλαδή είναι βαθυπερατό. Στις εισόδους εφαρμόζουμε ημιτονοειδή σήματα 100 mV, περιόδου 100 ms μέσω των πηγών V1 και V2 του τύπου VSIN. Στο analysis setup επιλέγουμε AC Sweep με 20 βήματα ανά δεκάδα (Pts/decade), δηλαδή σε λογαριθμική κλίμακα (decade), με αρχική συχνότητα 0,1 Hz και τελική 10 kHz. Στο εικονιζόμενο διάγραμμα παρουσιάζονται οι αντίστοιχες αποκρίσεις στο πεδίο συχνοτήτων.



## Παράδειγμα 2.3

### (Γραμμή μεταφοράς με διακριτά στοιχεία)

Στο παράδειγμα αυτό μελετάται η συμπεριφορά μιας γραμμής μεταφοράς εξομοιωμένη με διακεκριμένα στοιχεία αυτεπαγωγής (L) και χωρητικότητας (C). Στην είσοδο εφαρμόζεται ένας παλμός τάσης μέσω της πηγής V1. Στο analysis setup επιλέγουμε transient με βήμα 2 ns και τελικό χρόνο 600 ns.




Όπως φαίνεται και στην παραπάνω εικόνα του Probe ο παλμός διαδίδεται στη έξοδο της γραμμής με καθυστέρηση περίπου 100 ns σε σχέση με τον αρχικό. Στον αντιστάτη φόρτου δόθηκε η τιμή που αντιστοιχεί στη χαρακτηριστική αντίσταση μιας γραμμής μεταφοράς με καταναμημένα στοιχεία. Για το λόγο αυτό ο τερματισμός της γραμμής δεν είναι ιδανικός με αποτέλεσμα να παρουσιάζονται κυματώσεις που οφείλονται στις ανακλάσεις πάνω στον αντιστάτη φόρτου.





## Παράδειγμα 2.4 («Debouncer» με RS flip-flop)

Το παρακάτω κύκλωμα έχει σχεδιαστεί για να εφαρμόζει κανείς μοναδικούς παλμούς στην είσοδο δυαδικών μετρητών, αποτρέποντας πρόσθετους ανεπιθύμητους (παρασιτικούς) παλμούς. Αυτό επιτυγχάνεται με τη χρήση ενός R-S flip flop το οποίο αλλάζει κατάσταση στο κατερχόμενο μέτωπο των εισόδου του. Με τον τρόπο αυτό παρασιτικοί παλμοί ίδιας φοράς μεταβολής μετώπου δεν επιφέρουν καμιά αλλαγή κατάστασης, μέχρις ότου αναστραφεί η φορά μεταβολής. Οι αναστροφείς χρησιμοποιούνται για την εξομοίωση μεταγωγής εισόδου, δηλαδή όταν στην είσοδο Set δίνουμε 1, στην είσοδο Reset θα εφαρμοστεί 0 και αντιστρόφως. Στο εργαστήριο αυτό μπορεί να επιτευχθεί μέσω μεταγωγικού διακόπτη. Το δεύτερο κύκλωμα χρησιμοποιείται για αναλυθεί η ανάστροφη λειτουργία (Reset).





## Παράδειγμα 2.5 (πολυβάθμιος τελεστικός ενισχυτής)

Στο παράδειγμα αυτό γίνεται χρήση υποκυκλωμάτων που στην προκειμένη περίπτωση αφορούν δυο διαφορετικές συνδέσεις τελεστικών ενισχυτών, αναστρέφοντος ενισχυτή και ακολουθητή με το IC LF411. Παρατηρείστε τις εισόδους in-, out και in+, out αντίστοιχα, οι οποίες αντιστοιχούν πλήρως με αυτές που έχουν οριστεί στα εμφανιζόμενα υποκυκλώματα. Τα τρίγωνα στο εσωτερικό των πλαισίων των υποκυκλωμάτων δε σχετίζονται καθόλου με την προσομοίωση αλλά έχουν σχεδιαστεί για να συμβολίσουν τη λειτουργία τους. Η πηγή  $V_{in}$  είναι του τύπου VSIN, δηλαδή καθορισμένης επιλεγόμενης συχνότητας. Στο διάγραμμα απόκρισης που ακολουθεί το πλάτος εξόδου είναι πενταπλάσιο αυτού της εισόδου και σε φάση.






(δυναδικός μετρητής 4-bit)

Στο δυαδικό μετρητή του παραδείγματος αυτού χρησιμοποιείται το IC 74107A το οποίο περιέχει δύο J-K flip flops. Η συνδεσμολογία του κάθε flip flop χρησιμοποιείται ως υποκύκλωμα, ενώ οι αντιστάσεις είναι απαραίτητες για να γίνει η ανάλυση με τον τρόπο που γίνεται σε αναλογικά κυκλώματα. Στο analysis setup επιλέγουμε transient με κατάλληλο βήμα και τελικό χρόνο.



## Παράδειγμα 2.7 (τελεστικός ενισχυτής με offset)

Στο παράδειγμα αυτό χρησιμοποιούνται δύο βαθμίδες τελεστικού ενισχυτή. Η πρώτη δημιουργεί μια συνεχή τάση (offset) ενώ η δεύτερη αποτελείται από υποκύκλωμα το οποίο βασίζεται σε ένα μοντέλο τελεστικού ενισχυτή τύπου ακολουθητή. Το σήμα εισόδου είναι ημιτονοειδές που παρέχεται από την πηγή  $V_{in}$  τύπου VSIN.



## **ΒΙΒΛΙΟΓΡΑΦΙΑ - ΕΥΧΑΡΙΣΤΙΕΣ**

Οι Σημειώσεις αυτές είναι βασισμένες στα Βιβλία, COMPUTER-AIDED CIRCUIT ANALYSIS USING SPICE by Walter Banzhaf, Prentice Hall, 1989 και Introduction to PSpice by James W. Nilsson and Susan A. Riedel, Addison-Wesley, 1993. Επίσης, στα εγχειρίδια της MicroSim Corporation και άλλων εταιριών που παράγουν προϊόντα λογισμικού σχετικά με το SPICE και πληροφορίες που μπορούν να βρεθούν από διάφορες ιστοσελίδες στο Διαδίκτυο.

Η διαμόρφωση των Σημειώσεων έγινε με τέτοιο τρόπο ώστε αυτές να αποτελέσουν ένα εγχειρίδιο χρήσης του SPICE εμπλουτισμένο με θεωρητικά σχόλια, πρακτικούς κανόνες και εκτεταμένο πλήθος παραδειγμάτων. Σ' αυτό συνέβαλε η εμπειρία των συγγραφέων στη διδασκαλία του μαθήματος Ηλεκτρονικά και Εργαστήριο σε προπτυχιακό και μεταπτυχιακό επίπεδο.

Ευχαριστούμε τον κ. Δημήτρη Κουζή-Λουκά, Φοιτητή της Κατεύθυνσης Φυσικής της Σχολής Εφαρμοσμένων Μαθηματικών και Φυσικών Επιστημών του Ε.Μ. Πολυτεχνείου, ο οποίος, με τις γνώσεις του και την επιμελημένη εργασία του, έπαιξε σημαντικό ρόλο στη τελική έκδοση αυτών των σημειώσεων.